# Mathematical Quotients and Quotient Types in Coq

Laurent Chicli[1,2], Loïc Pottier[1], and Carlos Simpson[2]

[1] Projet Lemme, INRIA Sophia Antipolis
[2] Laboratoire J.A.Dieudonnè, Universitè de Nice Sophia-Antipolis

**Abstract.** This note studies quotient types in the Calculus of Inductive Constructions (CIC), implemented in the proof assistant `coq`, and compares their expressivity to that of mathematical quotients. In [Hof95], Martin Hofmann proposes an extension of the Calculus of Constructions (CC) with quotient types which he shows consistent, but notices that they are not sufficient to account for the natural isomorphism $\theta$ which exists in set theory between functional spaces $E \to \frac{F}{R}$ and $\frac{E \to F}{S}$ where $fSg$ iff $\forall x \in F,\ f(x)Rg(x)$. One can thus ask the question to know if it is possible to extend these quotient types to be able to show injectivity and surjectivity of this morphism. We show here that any extension of this kind in Coq with the impredicative sort `Set` would be contradictory.

## 1 Introduction

Let us consider two sets $E$ and $F$, $R$ an equivalence relation on $F$ and $\pi_R$ the canonical surjection from $F$ to $\frac{F}{R}$. We can define on $E \to F$ the following equivalence relation $S : fSg \Leftrightarrow \forall x \in E\ f(x)\ R\ g(x)$. By definition of the relation $S$, the map which composes any function $f : E \to F$ with $\pi_R$ is compatible with $S$, and there is thus a natural morphism $\theta$ such that the following diagram commutes:

$$\begin{tikzcd}
E \to F \arrow[r, "\pi_R \circ ."] \arrow[d, "\pi_S"] & E \to \dfrac{F}{R} \\
\dfrac{E \to F}{S} \arrow[ur, "\theta"] &
\end{tikzcd}$$

In set theory (where the axiom of choice and the extensionality of functions are given), this morphism is clearly injective (if $\pi_R\ o\ f = \pi_r\ o\ g$ i.e. $\forall x \in E,\ [f(x)]_R = [g(x)]_R$ one has $fSg$) and surjective (if $f$ is a function from $E$ to $\frac{F}{R}$, it is the image by $\theta$ of the function $\pi_S(g)$ where $g$ sends $x$ to a representative of the class of $f(x)$).

In the Calculus of Inductive Constructions implemented in `Coq` and extended by the quotient types of M.Hofmann [Hof95] the map $\theta$ can be defined (however

H. Geuvers and F. Wiedijk (Eds.): TYPES 2002, LNCS 2646, pp. 95–107, 2003.

we need the axiom of extensionality of functions) and can be shown injective, but not surjective. The goal of this paper is to show that one cannot extend these quotient types so as to obtain this surjectivity: the two types `E->F/R` and `(E->F)/S` are not and cannot be made isomorphic in the CIC of Coq, where impredicativity and strong elimination coexist for the sort `Set`.

The structure of this paper is the following. The problem is posed in details in section 2: one describes there the quotient types we consider, the difference in `Coq` between the proposition stating the existence of an object and the constructive data of it, finally we show the equivalence of the problem with the existence in general of a section for the canonical surjection of a quotient set ($\pi_R$).

In section 3, we show that the constructive data of a section is contradictory. We describe some results known on the quotients in CIC: the adaptation of the result of Diaconescu [Dia75] by Werner/Lacas [LacWer99] and results obtained by Maria Emilia Maietti [Mai98].

But our problem is equivalent to the *existence* of section for quotients and we show the inconsistency of this assumption in section 4. The first stage consists in showing the existence of a function `Proptobool` from `Prop` towards `bool` sending the true propositions on `true` and the false propositions on `false`. The second stage proceeds as follows: in a context where `False` (of type `Prop`) is the goal to prove, we can use elimination on the existence of `Proptobool` and thus use it to build an excluded middle in `Set`, which implies `False` ([Pot00], [Geu01]).

Using the same trick as in section 4, but independently of the concept of quotient type, we will see in section 5 another consequence of the impredicativity of `Set` : the excluded middle with the axiom of choice both in `Prop` lead to a contradiction in `Coq`.

All the proofs described here have been formalized in Coq V7.3[Coq02], and we give in this paper the script corresponding to the definitions, lemmas and proofs each time they are introduced[1].

## 2 Quotient Types in Coq

### 2.1 The Quotient Types

The quotient types that we consider were studied in [Hof95], [Bar95] and [Bou97]; they are a quasi-direct formalization of the mathematical practice of the quotients, from which we cut off, however, the concept of choice of a representative in an equivalence class.

For a type E and an equivalence relation R, they consist of:

- a type `quo` representing the quotient type,
- a function `class: E -> quo` for the canonical surjection,
- two propositions characterizing the equality of classes,
- a constructive way of descending to the quotient the functions compatible with $R$.

[1] One can find the complete scripts including proofs at the following URL: `http://www-sop.inria.fr/lemme/personnel/Laurent.Chicli/types2002.zip`.

– Finally, precisely because one has no function of choice of a representative in the equivalence classes, we give the surjectivity `quo_surj` of the class function stated in Prop .

More precisely, in Coq, we formalize these quotient types like this :

```
Record type_quotient [E : Type; R : (Relation E);
                      p : (Equivalence R)] : Type := {
   quo:> Type;
   class:> E->quo;
   quo_comp: (x,y:E)(R x y)->(class x) == (class y);
   quo_comp_rev: (x,y:E)(class x)==(class y)->(R x y);
   quo_lift: (F:Type)(f:E->F)(compatible R f)->quo->F;
   quo_lift_prop:
     (F:Type)(f:E->F)(H:(compatible R f))
     (x:E)((comp (quo_lift F f H) class) x)== (f x);
   quo_surj: (c:quo)(ExT x:E | c==(class x))}.

Axiom quotient :
   (E:Type) (R:(Relation E))(p:(Equivalence R))(type_quotient p).
```

Using these quotient types, we can define the equivalence relation S and the morphism $\theta$ in `Coq` (the extentionality of functions is necessary for that but is known to be consistent with the quotient types [Hof95]):

```
Variable E,F:Type.
Variable R:(Relation E); p:(Equivalence R).

Definition S:=[f,g:F->E](x:F)(R (f x) (g x)).

Lemma Sequiv: (Equivalence S).

Axiom Extensionality :
   (F,G:Type)(f,g:F->G)((x:F)(f x) == (g x))->f == g.

Lemma ps:(compatible S [f:F->E](comp (class (quotient p)) f)).

Definition theta:=(!quo_lift ? ? ? (quotient Sequiv) ? ? ps).
```

### 2.2 `Prop` and `Set`

The bottom of the hierarchy of universes of `Coq` consists of two kinds: `Prop` and `Set`. The first, `Prop`, is intended to represent the type of the logical propositions: a proposition in Coq is a type of type `Prop` . The second, `Set`, is intended for the *objects of the speech*: the set of natural numbers, the lists, the programs are generally built in `Set`.

Since Set's are intended to represent data-types, we want them to be non-degenerated (i.e. $0 \neq 1$.) To be able to prove that 0 differs from 1, one must allow (from a technical point of view) the elimination from `Set` to `Type`. The eliminations from Set to `Prop` or `Set` follow by cumulativity.

On the other hand, Prop's are propositions, that may be used either intuitionistically or classically. For the classical mathematician, the fact that proofs may be equal or different is irrelevant, but it is also true that many interesting 'classical' axioms imply the proof-irrelevance. For this reason, and to keep the possibility of classical reasoning, the system has to forbid any rule that could allow the user to prove that two proof-terms are different. This is why there is no elimination rule from `Prop` to `Set` or `Type` in `Coq` (otherwise, one could distinguish the two proofs of `True` $\vee$ `True` by case'ing them to the booleans).

The existence of an object $x$ verifying a property $P$ being thus defined in `Prop`:

```
Inductive exT [A : Type; P : A->Prop]  : Prop :=
      exT_intro : (x:A)(P x)->(ExT P)
```

we will be able to make an elimination on the existence of $x$ (and thus to have $x$ in our context) to build a term of kind `Prop` (the proof of any proposition) but not to build a term of kind `Set` (and thus an object). To affirm the existence of an object is thus weaker than to give it in `Set`: in the first case we will be able to make use of it only to show propositions, in the second case we will be able to go further and to build new objects with it.

In the continuation of this paper the difference between the two concepts is fundamental and each reference to the existence of an object will indicate only the proposition (in `Prop`) stating its existence.

On the other hand, "strong elimination" is authorized for inductive objects defined in `Set`, and this together with the impredicative nature shared by both `Set` and `Prop` is at the basis of the paradoxes which we use below.

### 2.3 The Existence of a Section for a Quotient Type

Let us return to our initial problem. In fact, for given $F$ and $R$, the surjectivity of the morphisms $\theta$ for all $E$ is equivalent to the existence of a section of $\pi_R$, i.e. a function $s$ such as

$$\exists s : \frac{F}{R} \to F \mid \pi_R \circ s = Id_{\frac{F}{R}}.$$

When $E = \frac{F}{R}$, the surjectivity of $\theta$ and $\pi_S$ implies that there exists $s : \frac{F}{R} \to F$ such that $\theta(\pi_S(s)) = Id_{\frac{F}{R}}$. Since the diagram of the first page commutes, we thus have $\pi_R \circ s = Id_{\frac{F}{R}}$, which expresses that $s$ is a section from $F/R$ to $F$.

Conversely, if there is a function $s : \frac{F}{R} \to F$ such as $\pi_R \circ s = Id_{\frac{F}{R}}$, then for all $E$ the corresponding map $\theta$ is surjective: for all $F : E \to \frac{F}{R}$ we have:

$$\theta(\pi_S(s \circ f)) = \pi_R \circ (s \circ f) = (\pi_R \circ s) \circ f = f.$$

In the continuation, only the first implication will interest us, and will give the following code in Coq :

```
Hypothesis theta_surj : (f : F ->  (quotient p))
  (EXT g : (quotient Sequiv) | (theta g) == f).
```

one shows:

```
Lemma step1: (EXT  s:(quo (quotient p))->E |
(comp (class (quotient p)) s) == (Id (quotient p))).
```

The surjectivity of the morphism $\theta$ is thus brought back to the question of the possibility of extending the quotient types by the existence of a section. In the next part, we will briefly recall results of [Pot00] where such a section is given constructively, and we will show the inconsistency of this assumption. It is only in the fourth section that we will treat the case where only the *existence* of the section is assumed.

# 3 To Give Constructively a Function of Choice Leads to a Contradiction

We are here within a framework where, in addition to our quotient types, a function of choice of representatives in the equivalence classes `s: quo -> E` is given, such that:

$$E \underset{s}{\overset{c}{\rightleftarrows}} quo \qquad \forall x : quo,\ (c(s\ x)) == x.$$

## 3.1 The Trick of Diaconescu

B.Werner and S.Lacas showed in [LacWer99] (by adapting the trick of Diaconescu [Dia75][2] ) that in this case, for a given proposition $P$, one can use the quotients to build a term of type `{P}+{~P}` deciding if $P$ is true or not. They proceed in the following way.

Suppose $E = bool$ and let us define the relation $R$ by:

$$x\ R\ y \Leftrightarrow x = y \vee P,$$

$R$ is indeed an equivalence relation, and if we consider the quotient $E/R$ (it depends of $P$), we have:

$$\begin{aligned}(s\ (c\ true)) =_{bool} (s\ (c\ false)) &\Leftrightarrow (c\ true) == (c\ false)\\ &\Leftrightarrow P\end{aligned}$$

And, the equality in `bool` being decidable, we obtain the decidability of $P$.

[2] Bridges in [Bri98] remarks that this trick is probably the one that Bishop had in mind as in exercise in [[Bish67],p.58,pb2] in 1967.

### 3.2 Inconsistency of the Excluded Middle in `Set`

By generalizing the Werner-Lacas construction for any proposition $P$ one can thus build a term of type `(P:Prop){P}+{~P}`, i.e. the strong excluded middle, in the kind `Set`.

To conclude, we precisely use the fact that the excluded middle in the kind `Set` is contradictory in `Coq`.

The interested reader can refer to [Pot00][3] for a demonstration of this inconsistency. There one takes as a starting point the result shown by Barbanera-Berrardi in [BarBer96] (the axiom of choice[4] with the excluded middle implies proof-irrelevance) and adapts it to `Set` (which is possible since `Set` is impredicative) in `Coq` showing that the strong excluded middle implies the equality of all the terms in a same type of kind `Set`, which is false (for example one can show by strong elimination that $0 \neq 1$ in the type `nat` of the natural numbers of `Coq`).

Another demonstration, more complicated, can be found in [Geu01], based on an adaptation of the Burali-Forti paradox (by Girard Coquand and Hurkens). This proof also uses in a strong way the impredicativity of `Set`.

### 3.3 Related Work

We can cite the work of Maria Emilia Maietti. In [Mai98], she shows that extending Martin-Löf type theory with quotient types (without section) and with the rule of uniqueness of equality proofs implies the excluded middle.

If we try to put this result in the CIC, we prove in `Coq`[5](with our quotient types, without section) that the axiom of choice with the rule of uniqueness of equality proofs implies the excluded middle in `Prop`.

However if we try to adapt this proof to get strong excluded middle, we have to use a strong axiom of choice, which implies our first hypothesis: the section $s$ constructively given.

## 4 Even the Existence of a Section Leads to a Contradiction

This time, the section `s: quo -> E` is not given anymore, but we have only the proposition ensuring its existence:

[3] This formalization supposes, in addition to the strong excluded middle, the axiom *eqT_rec* which makes it possible to rewrite on objects of kind `Set` if one has two equal terms for the Leibnitz equality in kind *Type*. Although noncontradictory with the CIC, this axiom was not originally included in `Coq` in order to preserve the mechanism of extraction. However, this qualification is no longer needed since the version 7.3 of `Coq` where *eqT_rec* was definitively integrated into the system.

[4] This particular axiom of choice, when formalized to show this paradox, is stated in Prop and for propositions, and is provable in `Coq` since elimination is authorized from `Prop` to `Prop`.

[5] The Coq script can be found at this url :
`http://www-sop.inria.fr/lemme/personnel/Laurent.Chicli/t02_rel_work.zip`

$$\begin{tikzcd} E \arrow[r, shift left, "c"] & quo \arrow[l, shift left, dashed, "s"] \end{tikzcd} \qquad \exists s : quo \to E,\ \forall x : quo,\ (c\ (s\ x)) == x.$$

We introduce here a new trick, as the old trick of Diaconescu does not work anymore (the corresponding explanations are postoned to subsection 4.3).

### 4.1 From `Prop` towards `Bool`

The goal of this section is to show the existence of a function `Proptobool: Prop -> bool` mapping the true propositions to `true` and false propositions to `false`, i.e. such that:

$$\forall P : Prop,\ P \leftrightarrow (Proptobool\ P) == true. \qquad (i)$$

**A Particular Quotient.** With this intention, we will use the existence of the section for the following particular quotient.

Thus let us consider the inductive type $P2$, made up of two copies of `Prop`

```
Inductive P2 : Type :=
   p1: Prop ->  P2
 | p2: Prop ->  P2 .
```

We define the following relation on the elements of $P2$ : all the equivalent propositions of the same copy of `Prop` are equivalent, as well as the true propositions of $p1$ and $p2$.




In `Coq` , that gives:

```
Inductive R : P2 -> P2 ->  Prop :=
   R_12: (x, y : Prop) x -> y ->  (R (p1 x) (p2 y))
 | R_21: (x, y : Prop) x -> y ->  (R (p2 x) (p1 y))
 | R_11: (x, y : Prop) (iff x y) -> (R (p1 x) (p1 y))
 | R_22: (x, y : Prop) (iff x y) -> (R (p2 x) (p2 y)) .

Lemma Requiv: (Equivalence R).
```

and we define thus the quotient type `QP2`:

```
Definition QP2 := (quotient Requiv).
```

**Some Definitions and Properties.** Since we want to define a term of kind `Prop` (we want to prove *the existence* of the function `Proptobool`), we can eliminate the existential quantifier on $s$ and have $s$ in our context.

Let us consider the function $f = s \circ c$, from $P2$ to itself. It has the following two properties:

– $\forall x : P2,\ xRf(x) :$ (1)

$$\begin{aligned}\forall x : P2,\ xR(fx) &\Leftrightarrow (c\ x) == (c\ (f\ x))\\ &\Leftrightarrow (c\ x) == (c\ (s\ (c\ x)))\\ &\Leftrightarrow (c\ x) == (c\ x)\end{aligned}$$

– $\forall x, y : P2,\ xRy \Leftrightarrow f(x) == f(y) :$ (2)

$$\begin{aligned}\forall x, y : P2,\ xRy &\Leftrightarrow (c\ x) == (c\ y)\\ &\Leftrightarrow (s\ (c\ x)) == (s\ (c\ y))\end{aligned}$$

Let us define the following functions `in_p1` and `in_p2`:

```
Definition in_p1: P2 ->  bool :=
   [x:P2]Cases x of (p1 p)=>true  | (p2 p)=> false end.

Definition in_p2 : P2 ->  bool :=
   [x:P2]Cases x of (p1 p)=>false | (p2 p)=> true end.
```

We have the following property : if an element $p_1(P)$ is in relation with an element $p_2(Q)$ then $P$ is true.

$$\forall P, Q : \mathrm{P}rop,\ p_1(P)\ R\ p_2(Q) \rightarrow P. \tag{3}$$

The proof is obvious, and is done easily in `Coq` by inversion on the hypothesis.

**The Existence of the Function `Proptobool`.** We now have all the necessary bricks to show the existence of the function `Proptobool`. We proceed by case analysis on `(in_p2 (f (p1 True)))` to define the witness ensuring the existence of the Proptobool function :

```
Exists Cases (in_p2 (f (p1 True))) of
          true  => [p : Prop] (in_p2 (f (p1 p)))
        | false => [p : Prop] (in_p1 (f (p2 p)))
       end.
```

Let us show that this function has the desired property $i$). In the first case, `(in_p2 (f (p1 True))) = true` and the function `Proptobool` is equal to $in\_p2 \circ f \circ p1\ :\ Prop \rightarrow bool.$




Let P be a proposition. If P is true, then :

$$P \leftrightarrow True$$

and thus

$$p1(P)\ R\ p1(True).$$

By (2), it follows that

$$f(p1(P)) == f(p1(True))$$

and thus:

$$Proptobool(P) == in\_p2(f(p1(P))) == in\_p2(f(p1(True))) == true.$$

Conversely, if $Proptobool(P) == in\_p_2(f(p_1(P))) == true$, we get by cas'ing $f(p_1(P))$ :

$$f(p_1(P)) == p_2(Q),$$

since the other case $f(p_1(P)) == p_1(Q)$ is absurd. We also have, by (1) :

$$p_1(P)\ R\ f(p_1(P)),$$

hence :

$$p_1(P)\ R\ p_2(Q),$$

and we get by (3) a proof of $P$.

In the second case, if `(in_p2 (f (p1 True))) = false` one sends this time `Prop` in `bool` through $in\_p1 \circ f \circ p2$, and we show in a similar way that it verifies the property $i$).



We show thus in Coq , without using any form of excluded middle:

```
Lemma existence_decision:
(EXT Proptobool:Prop->bool |
        (P:Prop)( P <-> (Proptobool P) = true)).
```

### 4.2 Inconsistency

The idea to show the inconsistency in this case is the same one as before: use Proptobool to show the excluded middle in Set, which is contradictory. However, we cannot show here (P:Prop){P}+{~P} using Proptobool since we do not have *a priori* the function itself but only its existence, and that the elimination of an object of Prop (the existential quantifier) is not possible when the kind of goal is Set what is the case here.

We circumvent the difficulty by noticing that what we want to show, at the end, is a proposition: False. We thus can, to build a term of the type False, make an elimination on the existence of Proptobool and thus introduce it in our context.

Then we just make a cut with (P:Prop)({P}+{~P}), which we easily prove for all P by case analysis on (Proptobool P).

```
Lemma incoherence: False.
Generalize existence_decision.
Intros yH.
Elim yH; Intros Proptobool H; Clear yH.
Cut (P:Prop){P}+{~P}.
Exact paradoxe.
Intros P.
Generalize (refl_equal ? (Proptobool P)).
Pattern -1 (Proptobool P); Case (Proptobool P).
Intros H0.
Case (H P).
Intros H1 H2.
Left; Exact (H2 H0).
Intros hyp; Right.
Unfold not;Intros p.
Case (H P).
Intros H0 H1.
Rewrite (H0 p) in hyp; Inversion hyp.
Qed.
```

### 4.3 Back to the Results of the Section 3

We can wonder whether the trick consisting in eliminating the existential quantifiers whereas the goal to prove is False makes useless any distinction between the constructive data of an object and the simple data of its existence.

Could we use, in our case, the constructions of Werner/Lacas/Diaconescu to show the inconsistency of the existence of the section? It seems that the answer is negative.

Indeed the construction of the quotient of the section 3.1 depends on a given proposition `P`, we thus build a term of type

$$H_0 : \quad \forall P : Prop,\ \forall s : E/R_P \to E,\ \forall x\ c(s(x)) == x \Rightarrow \{P\} + \{\tilde{}P\}.$$

In addition, we have in our context a proposition ensuring us, for all $P$, the existence of a section :

$$H_1 : \forall P : Prop,\ \exists s : E/R_P \to E \ \mid\ \forall x\ c(s(x)) == x.$$

And we want to use the inconsistency of the strong exluded middle :

$$H2 : ((P : Prop)\{P\} + \{\tilde{}P\}) \to False.$$

So, the current goal being `False` under the assumptions `H0`, `H1`, `H2`, there are two possibilities to continue the proof. Either we cut with `(P:Prop){P}+{~P}`, and then introduce `P`; but the new goal (`{P}+{~P}`) is now in `Set` and we cannot eliminate the existential quantifier of `(H1 P)` any more. Or we try first (before any cut) to eliminate the existential quantifier of `H1`, but in order to prove later `(P:Prop){P}+{~P}` we have to do that for all propositions, which is impossible.

As we see, constructions of section 4 make it possible to go further (and so prove the inconsistency) because they treat all the propositions with only one quotient set.

## 5 Inconsistency of the Axiom of Choice with the Excluded Middle in `Prop`

Independently of the concept of quotient type, the proofs of section 4 can be used to show that the axiom of choice with values in `Prop` is contradictory with the excluded middle (in `Prop`).

The axiom of (unique) choice that we consider here is as follows:

```
Axiom choice :
   (A,B : Type)(R:A->B->Prop)
   ((x : A)  (EXT y : B |  (R x y))) ->
   ((x : A) (y1, y2 : B) (R x y1) -> (R x y2) ->  y1 == y2) ->
   (EXT  f : (A -> B) | (x : A)  (R x (f x))).
```

It affirms, for two types $A$ and $B$, that if for all $x$ of $A$ there is single $y$ such as $(R\ x\ y)$ (where $R$ is a relation between elements of $A$ and $B$) then there exists a function $f : A \to B$ such that for all $x$ we have $(R\ x\ f(x))$.

With this axiom and the excluded middle in `Prop`, we can prove once again the existence of the function `Proptobool`. With this intention, we apply the axiom of choice to the relation

$$\begin{aligned} R : &\quad Prop \to bool \to Prop \\ &:= [P : Prop][b : bool]P \leftrightarrow b = true \end{aligned}$$

And we just have to prove

$$\forall P : Prop,\ \exists ! b : bool,\ P \leftrightarrow b = true$$

which is done easily with the excluded middle: it suffices to distinguish between $P$ or ˜$P$ and to choose `true` or `false` consequently.

```
Lemma existence_decision:
 (EXT f : Prop ->  bool |  (P : Prop)  (iff P (f P) == true)).
Apply (!AC ? ? [P : Prop] [b : bool]  (iff P b == true)).
Intros P; Elim (EM P).
Intros p; Exists true; Split; Intuition.
Intros p; Exists false; Split; Intuition.
Inversion H.
Intros x y1 y2.
Elim (EM x).
Intros xx H0 H;Elim H0;Elim H;Intros;Transitivity true;Intuition.
Case y1; Case y2; Intuition.
Elim (H (H3 (refl_eqT bool true))).
Qed.
```

We then build a term of type `False` starting from the existence of `Proptobool` by taking again the proof of section 4.2.

## 6 Conclusion

We thus showed by using a single quotient type, that even the proposition stating the existence of a section for the canonical surjection of a quotient, leads to a contradiction in `Coq`, and thus that the morphism $\theta$ is not, in general, surjective in the CIC with quotient types. In this direction the expressivity of the quotient types in `Coq` seems definitively weaker than the one of the mathematical quotients.

It is the kind `Set` which is problematic, and more precisely its impredicativity: indeed, the impredicativity of `Set` is essential in the proof of the inconsistency of `(P:Prop){P}+{~P}` and without it, the classical boolean model makes possible to show that the surjectivity of $\theta$ is not contradictory any more. One of the solutions to allow `Coq` to have quotient types with the same properties on functional spaces as the mathematical quotients would be thus to extinguish the impredicativity of `Set`.

Let us finish by giving two alternatives to quotient types to represent quotient sets in `Coq`. The first one consists in representing sets not as a type with Leibnitz equality but as a type with an equivalence relation : a setoïd. So a quotient set is simply a setoid with another equivalence relation than equality. It is currently the most used solution, although it's very tiresome to work with. The second possibility is to use the work of Pierre Courtieu [Cou01] : quotients with computable "canonical form", to have quotient types with constructive choice, but the class of the quotients we can thus formalize is restricted.

Finally, we would like to thank the referees for their corrections and suggestions.

## References


[BarBer96] F. Barbanera and S. Berardi. *Proof-Irrelevance out of Exluded-Middle and Choice in the Calculus of Constructions.* Journal of Functional Programming, pages 519–525, 1996.

[Bar95] G. Barthe. *Extensions of pure type systems.* In M. Dezani-Ciancaglini and G. Plotkin, editors, Proceedings of TLCA'95, volume 902, pages 16–31. Springer-Verlag, 1995.

[Bish67] E.Bishop. *Foundations of Constructive Analysis* McGraw-Hill, 1967.

[Bou97] S. Boutin. *Réflexions sur les quotients.* PhD thesis, Thèse de doctorat, Université de Paris 7, 1997.

[Bri98] Constructive truth in parctice. In H.G. Dales and G.Oliveri, editors, *Truth in Mathmatics.* Oxford University Press, 1998.

[Coq02] The Coq Development Team. *The Coq proof assistant V7.3 Reference Manual* `ftp://ftp.inria.fr/INRIA/coq/V7.3.1/doc`

[Cou01] P. Courtieu. *Représentation d'algèbres non libres en théorie des types.* PhD thesis, Thèse de doctorat, Université Paris-Sud, 2001.

[Dia75] R. Diaconescu. *Axiom of choice and complementation.* in Proceedings of A.M.S., vol. 51, pages 176-178, 1975.

[Geu01] H. Geuvers. *Inconsistency of classical logic in type theory.* Note. `http://www.cs.kun.nl/ herman/note.ps.gz`

[Hof95] M. Hofmann. *Extensional concepts in intensional type theory.* PhD thesis, LFCS Edinburgh, 1995.

[LacWer99] S. Lacas and B. Werner. *Which choices imply the Excluded Middle?* preprint, 1999.

[Mai98] Maria Emilia Maietti. *About Effective Quotients in Constructive Type Theory.* TYPES'98, LNCS 1657,pp 164-178,1999.

[Pot00] Loïc Pottier. *Quotients dans le CCI.* Rapport de Recherche INRIA RR-4053. `http://www-sop.inria.fr/rapports/sophia/RR-4053.html`