

Some constructions on ω -groupoids

Thorsten Altenkirch
University of Nottingham
txa@cs.nott.ac.uk

Nuo Li
University of Nottingham
nzl@cs.nott.ac.uk

Ondřej Rypáček
University of Oxford
ondrej.rypacek@gmail.com

ABSTRACT

Weak ω -groupoids are the higher dimensional generalisation of setoids and are an essential ingredient of the constructive semantics of Homotopy Type Theory [9]. Following up on our previous formalisation [3] and Brunerie’s notes [6], we present a new formalisation of the syntax of weak ω -groupoids in Agda using heterogeneous equality. We show how to recover basic constructions on ω -groupoids using suspension and replacement. In particular we show that any type forms a groupoid and that we can derive higher dimensional composition. We present a possible semantics using globular sets and discuss the issues which arise when using globular types instead.

Categories and Subject Descriptors

F.4.1 [Mathematical Logic and Formal Languages]: Lambda calculus and related systems, Mechanical theorem proving

General Terms

Theory

Keywords

Type Theory, Homotopy Type Theory, Category Theory, Higher dimensional structures

1. INTRODUCTION

In Type Theory, a type can be interpreted as a setoid which is a set equipped with an equivalence relation [1]. The equivalence proof of the relation consists of reflexivity, symmetry and transitivity whose proofs are unique. However in Homotopy Type Theory, we reject the principle of uniqueness of identity proofs (UIP). Instead we accept the univalence axiom which says that equality of types is weakly equivalent to weak equivalence. Weak equivalence can be seen as a refinement of isomorphism without UIP [3]. For example, a weak equivalence between two objects A and B in a

2-category is a morphism $f : A \rightarrow B$ which has a corresponding inverse morphism $g : B \rightarrow A$, but instead of the proofs of isomorphism $f \circ g = 1_B$ and $g \circ f = 1_A$ we have two 2-cell isomorphisms $f \circ g \cong 1_B$ and $g \circ f \cong 1_A$.

Voevodsky proposed the univalence axiom which basically says that isomorphic types are equal. This can be viewed as a strong extensionality axiom and it does imply functional extensionality (a coq proof of this can be found in [5]). However, adding univalence as an axiom destroys canonicity, i.e. that every closed term of type \mathbb{N} is reducible to a numeral. In the special case of extensionality and assuming a strong version of UIP we were able to eliminate this issue [1, 2] using setoids. However, it is not clear how to generalize this in the absence of UIP to univalence which is incompatible with UIP. To solve the problem we should generalise the notion of setoids, namely to enrich the structure of the identity proofs.

The generalised notion is called weak ω -groupoids and was proposed by Grothendieck 1983 in a famous manuscript *Pursuing Stacks* [7]. Maltsiniotis continued his work and suggested a simplification of the original definition which can be found in [8]. Later Ara also presents a slight variation of the simplification of weak ω -groupoids in [4]. Categorically speaking an ω -groupoid is an ω -category in which morphisms on all levels are equivalences. As we know that a set can be seen as a discrete category, a setoid is a category where every morphism is unique between two objects. A groupoid is more generalised, every morphism is isomorphism but the proof of isomorphism is unique, namely the composition of a morphism with its inverse is equal to an identity morphism. Similarly, an n -groupoid is an n -category in which morphisms on all levels are equivalence. ω -groupoids which are also called ∞ -groupoids is an infinite version of n -groupoids. To model Type Theory without UIP we also require the equalities to be non-strict, in other words, they are not definitionally equalities. Finally we should use weak ω -groupoids to interpret types and eliminate the univalence axiom.

There are several approaches to formalise weak ω -groupoids in Type Theory. For instance, Altenkirch and Rypáček [3], and Brunerie’s notes [6]. This paper mainly explains an implementation of weak ω -groupoids following Brunerie’s approach in Agda which is a well-known theorem prover and also a variant of intensional Martin-Löf type theory. **This is the first attempt to formalise this approach in Agda, and other languages like Coq.** The approach is to specify when

a globular set is a weak ω -groupoids by first defining a type theory called $\mathcal{T}_{\infty\text{-groupoid}}$ to describe the internal language of Grothendieck weak ω -groupoids, then interpret it with a globular set and a dependent function. All coherence laws of the weak ω -groupoids should be derivable from the syntax, we will present some basic ones, for example reflexivity. One of the main contributions of this paper is to use the heterogeneous equality for terms to overcome some very difficult problems when we used the normal homogeneous one. In this paper, we omit some complicated and less important programs, namely the proofs of some lemmas or the definitions of some auxiliary functions. It is still possible for the reader who is interested in the details to check the code online, in which there are only some minor differences.

Acknowledgements

The first and second author would like to thank the organizers and other participants of the special year on homotopy type theory at the Institute for Advanced Study where they had many interesting discussion topics related to the work presented in this paper. Especially we all would like to acknowledge Guillaume Brunerie's proposal which made this work possible.

2. SYNTAX

We develop the type theory of ω -groupoids formally, following [6]. This is a Type Theory with only one type former which we can view as equality types and interpret as the homsets of the ω -groupoid. There are no definitional equalities which correspond to the fact that we consider weak ω -groupoids. None of the groupoid laws on any levels are strict (i.e. definitional) but all are witnessed by terms. Compared to [3] the definition is very much simplified by the observation that all laws of a weak ω -groupoid follow from the existence of coherence constants for any contractible context.

In our formalisation we exploit the more liberal way to do mutual definitions in Agda, which was implemented recently following up a suggestion by the first author. It allows us to first introduce a type former but give its definition later.

Since we are avoiding definitional equalities we have to define a syntactic substitution operation which we need for the general statement of the coherence constants. However, defining this constant requires us to prove a number of substitution laws at the same time. We address this issue by using a heterogeneous equality which exploits uniqueness of identity proofs (UIP). Note that UIP holds because all components defined here are sets in the sense of Homotopy Type Theory.

2.1 Basic Objects

We first declare the syntax of our type theory which is called $\mathcal{T}_{\infty\text{-groupoid}}$ namely the internal language of weak ω -groupoids. The following declarations in order are contexts as sets, types are sets dependent on contexts, terms and variables are sets dependent on types, Contexts morphisms and the contractible contexts.

```
data Con      : Set
```

```
data Ty (Γ : Con) : Set
data Tm          : {Γ : Con} (A : Ty Γ) → Set
data Var         : {Γ : Con} (A : Ty Γ) → Set
data _⇒_         : Con → Con → Set
data isContr     : Con → Set
```

It is possible to complete the definition of contexts and types first. Contexts are inductively defined as either an empty context or a context with a type of it. Types are defined as either $*$ which we call it 0-cell, or a morphism between two terms of some type A. If the type A is n-cell then we call the morphism $(n + 1)$ -cell.

```
data Con where
  ε      : Con
  _-_-   : (Γ : Con) (A : Ty Γ) → Con

data Ty Γ where
  *      : Ty Γ
  _=h_   : {A : Ty Γ} (a b : Tm A) → Ty Γ
```

2.2 Heterogeneous Equality for Terms

One of the big challenge we encountered at first is the difficulty to formalise and to reason about the equalities of terms, which is essential when defining substitution. When we used the common identity types which are homogeneous, we had to use *subst* function in Agda to unify the types on both sides of the equation. It created a lot of technical issues that made the encoding too involved to proceed.

A Case of complexity: For example, assume we have a context Γ , three types $A B : Ty \Gamma$ and we know $p : A \equiv B$. If we have a term $a : Tm A$ and a term $b : Tm B$, it is impossible to just write $a \equiv b$ because of type unification. We need to write $substTm p a \equiv b$. Again assume we have another type $C : Ty \Gamma$, $q : B \equiv C$ and a term of $c : Tm C$, to prove a possible lemma $a = c$, we have to write something like $subst Tm q (subst Tm p a) \equiv c$. Of course due to this possible nesting of "subst" inside another "subst", we need to prove a coherence lemma $subst Tm q (subst Tm p a) \equiv subst Tm (trans p q) a$ to help us in other proofs. Indeed, there are much more coherence lemmas to prove to deal with the *subst* issues.

However we found that the equality of syntactic types of given context which will be introduced later, is decidable which means that it is an h-set. In other words, the equality of syntactic types is unique, so that it is safe to use the JM equality (heterogeneous equality) for terms of different types. The heterogeneous equality for terms is inhabited only when their syntactic types are definitionally equal.

In intensional type theory, the uniqueness of identity proof (UIP) in general is not provable, namely not all types are h-sets (homotopy 0-type). However it is justified to claim some types are h-sets. From the Hedberg Theorem we know that inductive types with finitary constructors is a set. Here these types which stands for syntactic components are inductive-inductive types with finitary constructors, and from the same reason, it is safe to claim the UIP for the syntactic contexts,

types, terms and variables are all sets. Here we only the UIP for syntactic types so that we can adopt heterogeneous equality for terms which can simplify the work a lot if it is safe. The motivation of using the heterogeneous equality is due to the complexity when manipulating equality of terms. Without the heterogeneous equality, most propositional equality of terms involves a lot of *subst* (which substitute certain values in the dependent type of terms), and repetitive utilization of the coherence laws for *subst*.

```
data _≅_ {Γ : Con} {A : Ty Γ} :
  {B : Ty Γ} → Tm A → Tm B → Set where
  refl : (b : Tm A) → b ≅ b
```

Once we have the heterogeneous equality for terms, we can define a proof-irrelevant substitution which we call coercion here since it gives us a term of type A if we have a term of type B and the two types are equal. We can also prove that the coerced term is heterogeneously equal to the original term. Combining these definitions, it is much more convenient to formalise and to reason about term equations.

```
_[]_ : {Γ : Con} {A B : Ty Γ} {a : Tm B}
      → A ≡ B → Tm A
a [] refl = a

cohOp : {Γ : Con} {A B : Ty Γ} {a : Tm B} (p : A ≡ B)
      → a [] p ≅ a
cohOp refl = refl _
```

2.3 Substitutions

With context morphism, we can define substitutions for types variables and terms. Usually we define a set of symbols together and we name a function $*$ as $*C$ for contexts, $*T$ for types, $*V$ for variables $*tm$ for terms and $*S$ for context morphisms. For example the substitution for types is defined as follows

```
_[]T : {Γ Δ : Con} → Ty Δ → (δ : Γ ⇒ Δ) → Ty Γ
```

In this paper, we will not present the others in the same set, but you can find them in the code.

Indeed the composition of contexts can be understood as substitution for context morphisms as well.

```
_@_ : {Γ Δ Θ : Con} → Δ ⇒ Θ → (δ : Γ ⇒ Δ) → Γ ⇒ Θ
```

2.4 Weakening Rules

We can freely add types to the contexts of any given type judgments, term judgments or context morphisms. These are weakening rules.

```
_+T_ : {Γ : Con} (A : Ty Γ) → (B : Ty Γ) → Ty (Γ , B)
```

To define the variables we have to use the weakening rules. We use typed de Bruijn indices to define variables as either the rightmost variable of the context, or some variable in the context which can be found by cancelling the rightmost variable along with each vS . The coherence constants are one of the major part of this syntax, which are primitive terms of the primitive types in contractible contexts which will be introduced later. Since contexts, types, variables and terms are all mutually defined, most of their properties have to be proved simultaneously.

```
data Var where
  v0 : {Γ : Con} {A : Ty Γ} → Var (A +T A)
  vS : {Γ : Con} {A B : Ty Γ} (x : Var A) → Var (A +T B)
```

A term can be either a variable or a coherence constant (*coh*). It encodes all constants for arbitrary types in a contractible context.

```
data Tm where
  var : {Γ : Con} {A : Ty Γ} → Var A → Tm A
  coh : {Γ Δ : Con} → isContr Δ → (δ : Γ ⇒ Δ)
      → (A : Ty Δ) → Tm (A [] δ T)
```

With variables defined, it is possible to formalise another core part of the syntactic framework, *contractible contexts*. Intuitively speaking, a context is contractible if its geometric realization is contractible to a point. It either contains one variable of the 0-cell $*$ which is the base case, or we can extend a contractible context with a variable of an existing type and an n-cell, namely a morphism, between the new variable and some existing variable. The graph can be drawn like branching trees.

```
data isContr where
  c* : isContr (ε , *)
  ext : {Γ : Con} → isContr Γ → {A : Ty Γ} (x : Var A)
      → isContr (Γ , A , (var (vS x) =h var v0))
```

Context morphisms are defined inductively similarly to contexts. A context morphism is a list of terms corresponding to the list of types in the context on the right hand side of the morphism.

```
data _⇒_ where
  • : {Γ : Con} → Γ ⇒ ε
  -.- : {Γ Δ : Con} (δ : Γ ⇒ Δ) {A : Ty Δ}
      (a : Tm (A [] δ T)) → Γ ⇒ (Δ , A)
```

2.5 Lemmas

The following lemmas are essential for constructions and theorem proving in later part. The first set of lemmas state that to substitute a type, a variable, a term, or a context morphism with two context morphisms consecutively, is equivalent to substitute with the composition of the two

context morphisms:

$$\begin{aligned} [\odot]T & : \{ \Gamma \Delta : \mathbf{Con} \} \{ A : \mathbf{Ty} \} \{ \Theta : \Delta \Rightarrow \Theta \} \\ & \{ \delta : \Gamma \Rightarrow \Delta \} \rightarrow A \ [\vartheta \odot \delta] T \equiv (A \ [\vartheta] T) [\delta] T \end{aligned}$$

The second set state that weakening inside substitution is equivalent to weakening outside:

$$\begin{aligned} [+S]T & : \{ \Gamma \Delta : \mathbf{Con} \} \{ A : \mathbf{Ty} \} \{ \delta : \Gamma \Rightarrow \Delta \} \{ B : \mathbf{Ty} \} \Gamma \\ & \rightarrow A \ [\delta +S B] T \equiv (A \ [\delta] T) +T B \end{aligned}$$

TO DO: maybe hide this part There are also some auxiliary functions derived from these lemmas. For instance, the function shown below is used a lot in proofs.

$$\begin{aligned} \mathbf{wk-tm+} & : \{ \Gamma \Delta : \mathbf{Con} \} \{ A : \mathbf{Ty} \} \{ \delta : \Gamma \Rightarrow \Delta \} \{ B : \mathbf{Ty} \} \Gamma \\ & \rightarrow \mathbf{Tm} (A \ [\delta] T +T B) \rightarrow \mathbf{Tm} (A \ [\delta +S B] T) \\ \mathbf{wk-tm+} \ B \ t & = t \ [[+S]T] \end{aligned}$$

TO DO: maybe hide this part

We can cancel the last term in the substitution for weakened objects since weakening doesn't introduce new variables in types and terms.

$$\begin{aligned} +T[.]T & : \{ \Gamma \Delta : \mathbf{Con} \} \{ A : \mathbf{Ty} \} \{ \delta : \Gamma \Rightarrow \Delta \} \\ & \{ B : \mathbf{Ty} \} \{ b : \mathbf{Tm} (B \ [\delta] T) \} \\ & \rightarrow (A +T B) \ [\delta , b] T \equiv A \ [\delta] T \end{aligned}$$

Most of the substitutions are defined as usual, except the one for coherence constants. We do substitution in the context morphism part of the coherence constants.

$$\begin{aligned} \mathbf{var} \ x & \quad [\delta] \mathbf{tm} = x \ [\delta] V \\ \mathbf{coh} \ c\Delta \ \gamma \ A & \quad [\delta] \mathbf{tm} = \mathbf{coh} \ c\Delta \ (\gamma \odot \delta) \ A \ [\mathbf{sym} [\odot]T] \end{aligned}$$

3. SOME IMPORTANT DERIVABLE CONSTRUCTIONS

In this section we show that it is possible to reconstruct the structure of a (weak) ω -groupoid from the syntactical framework presented in Section 2 in the style of [3]. To this end, let us call a term $a : \mathbf{Tm} \ A$ an n -cell if $\mathbf{level} \ A \equiv n$, where

$$\begin{aligned} \mathbf{level} & : \forall \{ \Gamma \} \rightarrow \mathbf{Ty} \ \Gamma \rightarrow \mathbb{N} \\ \mathbf{level} \ * & = 0 \\ \mathbf{level} \ (_ =h_ \{ A \} _) & = \mathbf{suc} \ (\mathbf{level} \ A) \end{aligned}$$

In any ω -category, any n -cell a has a domain (source), $s_m^n a$, and a codomain (target), $s_m^n a$, for each $m \leq n$. These are,

of course, $(n-m)$ -cells. For each pair of n -cells such that for some m $s_m^n a \equiv t_m^n b$, there must exist their composition $a \circ_m^n b$ which is an n -cell. Composition is (weakly) associative. Moreover for any $(n-m)$ -cell x there exists an n -cell $\mathbf{id}_m^n x$ which behaves like a (weak) identity with respect to \circ_m^n . For the time being we discuss only the construction of cells and omit the question of coherence.

For instance, in the simple case of bicategories, each 2-cell a has a horizontal source $s_1^1 a$ and target $t_1^1 a$, and also a vertical source $s_1^2 a$ and target $t_1^2 a$, which is also the source and target, of the horizontal source and target, respectively, of a .

There is horizontal composition of 1-cells $\circ_1^1 : x \xrightarrow{f} y \xrightarrow{g} z$, and also horizontal composition of 2-cells \circ_1^2 , and vertical composition of 2-cells \circ_2^2 . There is a horizontal identity on a , $\mathbf{id}_1^1 a$, and vertical identity on a , $\mathbf{id}_1^2 a = \mathbf{id}_2^1 \mathbf{id}_1^1 a$.

Thus each ω -groupoid construction is defined with respect to a *level*, m , and depth $n-m$ and the structure of an ω -groupoid is repeated on each level. As we are working purely syntactically we may make use of this fact and define all groupoid structure only at level $m = 1$ and provide a so-called *replacement operation* which allows us to lift any cell to an arbitrary type A . It is called 'replacement' because we are syntactically replacing the base type $*$ with an arbitrary type, A .

An important general mechanism we rely on throughout the development follows directly from the type of the only non-trivial constructor of \mathbf{Tm} , \mathbf{coh} , which tells us that to construct a new term of type $\Gamma \vdash A$, we need a contractible context, Δ , a type $\Delta \vdash T$ and a context morphism $\delta : \Gamma \Rightarrow \Delta$ such that

$$T[\delta] T \equiv A$$

Because in a contractible context all types are inhabited we may in a way work freely in Δ and then pull back all terms to A using δ . To show this formally, we must first define identity context morphisms which complete the definition of a *category* of contexts and context morphisms:

$$\mathbf{IdCm} : \forall \{ \Gamma \} \rightarrow \Gamma \Rightarrow \Gamma$$

It satisfies the following property:

$$\mathbf{IC-T} : \forall \{ \Gamma : \mathbf{Con} \} \{ A : \mathbf{Ty} \} \Gamma \rightarrow A \ [\mathbf{IdCm}] T \equiv A$$

The definition proceeds by structural recursion and therefore extends to terms, variables and context morphisms with analogous properties. It allows us to define at once:

$$\begin{aligned} \mathbf{Coh-Contr} & : \{ \Gamma : \mathbf{Con} \} \{ A : \mathbf{Ty} \} \Gamma \rightarrow \mathbf{isContr} \ \Gamma \rightarrow \mathbf{Tm} \ A \\ \mathbf{Coh-Contr} \ isC & = \mathbf{coh} \ isC \ \mathbf{IdCm} \ _ [\mathbf{sym} \ \mathbf{IC-T}] \end{aligned}$$

We use $\mathbf{Coh-Contr}$ as follows: for each kind of cell we want to define, we construct a minimal contractible context built out of variables together with a context morphism that pop-

ulates the context with terms and a lemma that states a definitional equality between the substitution and the original type.

3.1 Suspension and Replacement

For an arbitrary type A in Γ of level n one can define a context with $2n$ variables, called the *stalk* of A . Moreover one can define a morphism from Γ to the stalk of A such that its substitution into the maximal type in the stalk of A gives back A . The stalk of A depends only on the level of A , the terms in A define the substitution. Here is an example of stalks of small levels: ε (the empty context) for $n = 0$; $(x_0 : *, x_1 : *)$ for $n = 1$; $(x_0 : *, x_1 : *, x_2 : x_0 =_h x_1, x_3 : x_0 =_h x_1)$ for $n = 2$, etc.

This is the $\Delta = \varepsilon$ case of a more general construction where in we *suspend* an arbitrary context Δ by adding $2n$ variables to the beginning of it, and weakening the rest of the variables appropriately so that type $*$ becomes $x_{2n-2} =_h x_{2n-1}$. A crucial property of suspension is that it preserves contractibility.

3.1.1 Suspension

Suspension is defined by iteration level- A -times the following operation of one-level suspension. ΣC takes a context and gives a context with two new variables of type $*$ added at the beginning, and with all remaining types in the context suspended by one level.

$$\begin{aligned} \Sigma C &: \text{Con} \rightarrow \text{Con} \\ \Sigma T &: \{\Gamma : \text{Con}\} \rightarrow \text{Ty } \Gamma \rightarrow \text{Ty } (\Sigma C \Gamma) \\ \Sigma C \varepsilon &= \varepsilon, *, * \\ \Sigma C (\Gamma, A) &= \Sigma C \Gamma, \Sigma T A \end{aligned}$$

The rest of the definitions is straightforward by structural recursion. In particular we suspend variables, terms and context morphisms:

$$\begin{aligned} \Sigma v &: \{\Gamma : \text{Con}\} \{A : \text{Ty } \Gamma\} \rightarrow \text{Var } A \rightarrow \text{Var } (\Sigma T A) \\ \Sigma tm &: \{\Gamma : \text{Con}\} \{A : \text{Ty } \Gamma\} \rightarrow \text{Tm } A \rightarrow \text{Tm } (\Sigma T A) \\ \Sigma s &: \{\Gamma \Delta : \text{Con}\} \rightarrow \Gamma \Rightarrow \Delta \rightarrow \Sigma C \Gamma \Rightarrow \Sigma C \Delta \end{aligned}$$

The following lemma establishes preservation of contractibility by one-step suspension:

$$\Sigma C\text{-Contr} : (\Delta : \text{Con}) \rightarrow \text{isContr } \Delta \rightarrow \text{isContr } (\Sigma C \Delta)$$

It is also essential that suspension respects weakening and substitution:

$$\begin{aligned} \Sigma T[+T] &: \{\Gamma : \text{Con}\} \{A : \text{Ty } \Gamma\} \{B : \text{Ty } \Gamma\} \\ &\rightarrow \Sigma T (A + T B) \equiv \Sigma T A + T \Sigma T B \\ \Sigma tm[+tm] &: \{\Gamma : \text{Con}\} \{A : \text{Ty } \Gamma\} \{a : \text{Tm } A\} \{B : \text{Ty } \Gamma\} \\ &\rightarrow \Sigma tm (a + tm B) \equiv \Sigma tm a + tm \Sigma T B \end{aligned}$$

$$\begin{aligned} \Sigma T[\Sigma s]T &: \{\Gamma \Delta : \text{Con}\} \{A : \text{Ty } \Delta\} (\delta : \Gamma \Rightarrow \Delta) \\ &\rightarrow (\Sigma T A) [\Sigma s \delta]T \equiv \Sigma T (A [\delta]T) \end{aligned}$$

General suspension to the level of a type A is defined by iteration of one-level suspension. For symmetry and ease of reading the following suspension functions take as a parameter a type A in Γ , while they depend only on its level.

$$\begin{aligned} \Sigma C\text{-it} &: \{\Gamma : \text{Con}\} \{A : \text{Ty } \Gamma\} \rightarrow \text{Con} \rightarrow \text{Con} \\ \Sigma T\text{-it} &: \{\Gamma \Delta : \text{Con}\} \{A : \text{Ty } \Gamma\} \rightarrow \text{Ty } \Delta \rightarrow \text{Ty } (\Sigma C\text{-it } A \Delta) \\ \Sigma tm\text{-it} &: \{\Gamma \Delta : \text{Con}\} \{A : \text{Ty } \Gamma\} \{B : \text{Ty } \Delta\} \\ &\rightarrow \text{Tm } B \rightarrow \text{Tm } (\Sigma T\text{-it } A B) \end{aligned}$$

Finally, it is clear that iterated suspension preserves contractibility.

$$\begin{aligned} \Sigma C\text{-it-Contr} &: \forall \{\Gamma \Delta\} \{A : \text{Ty } \Gamma\} \rightarrow \text{isContr } \Delta \\ &\rightarrow \text{isContr } (\Sigma C\text{-it } A \Delta) \end{aligned}$$

By suspending the minimal contractible context, $*$, we obtain a so-called span. They are also stalks with a top variable added. For example $(x_0 : *)$ (the one-variable context) for $n = 0$; $(x_0 : *, x_1 : *, x_2 : x_0 =_h x_1)$ for $n = 1$; $(x_0 : *, x_1 : *, x_2 : x_0 =_h x_1, x_3 : x_0 =_h x_1, x_4 : x_2 =_h x_3)$ for $n = 2$, etc. Spans play an important role later in the definition of composition.

3.1.2 Replacement

After we have suspended a context by inserting an appropriate number of variables, we may proceed to a substitution which fills the stalk for A with A . The context morphism representing this substitution is called *filter*. In the final step we combine it with Γ , the context of A . The new context contains two parts, the first is the same as Γ , and the second is the suspended Δ substituted by *filter*. However, we also have to drop the stalk of A because it already exists in Γ .

Geometrically speaking, the context resulting from replacing $*$ in Δ by A is a new context which corresponds to the pasting of Δ to Γ to A .

As always, we define replacement for contexts, types and terms:

$$\begin{aligned} \text{rpl-C} &: \{\Gamma : \text{Con}\} \{A : \text{Ty } \Gamma\} \rightarrow \text{Con} \rightarrow \text{Con} \\ \text{rpl-T} &: \{\Gamma \Delta : \text{Con}\} \{A : \text{Ty } \Gamma\} \rightarrow \text{Ty } \Delta \rightarrow \text{Ty } (\text{rpl-C } A \Delta) \\ \text{rpl-tm} &: \{\Gamma \Delta : \text{Con}\} \{A : \text{Ty } \Gamma\} \{B : \text{Ty } \Delta\} \rightarrow \text{Tm } B \\ &\rightarrow \text{Tm } (\text{rpl-T } A B) \end{aligned}$$

Replacement for contexts, rpl-C , defines for a type A in Γ and another context Δ a context which begins as Γ and follows by each type of Δ with $*$ replaced with (pasted onto) A . To this end we must define the substitution *filter* which

pulls back each type from suspended Δ to the new context.

```

filter : {Γ : Con}(Δ : Con)(A : Ty Γ)
  → rpl-C A Δ ⇒ ΣC-it A Δ

rpl-C {Γ} A ε = Γ
rpl-C A (Δ , B) = rpl-C A Δ , rpl-T A B

rpl-T A B = ΣT-it A B [ filter _ A ]T

```

3.2 First-level Groupoid Structure

We can proceed to the definition of the groupoid structure of the syntax. We start with the base case: 1-cells. Replacement defined above allows us to lift this structure to an arbitrary level n (we leave most of the routine details out). This shows that the syntax is a 1-groupoid on each level. In the next section we show how also the higher-groupoid structure can be defined.

We start by an essential lemma which formalises the discussion at the beginning of this Section: to construct a term in a type A in an arbitrary context, we first restrict attention to a suitable contractible context Δ and use lifting and substitution – replacement – to pull the term built by `coh` in Δ back. This relies on the fact that a lifted contractible context is also contractible, and therefore any type lifted from a contractible context is also inhabited.

```

Coh-rpl : {Γ Δ : Con}(A : Ty Γ)(B : Ty Δ) → isContr Δ
  → Tm {rpl-C A Δ} (rpl-T A B)
Coh-rpl {Δ = Δ} A B isc =
  coh (ΣC-it-ε-Contr A isc) (filter Δ A) (ΣT-it A B)

```

Next we define the reflexivity, symmetry and transitivity terms of any type.

Reflexivity Let's start from the basic case as for the base type $*$. It is trivially inhabited because the context is the basic case of a contractible context.

```

refl* : Tm {x:*} (var v0 =h var v0)
refl* = Coh-Contr c*

```

To obtain the reflexivity term for any given type, we just use replacement.

```

refl-Tm : {Γ : Con}(A : Ty Γ)
  → Tm (rpl-T {Δ = x:*} A (var v0 =h var v0))
refl-Tm A = rpl-tm A refl*

```

Symmetry Symmetry (inverse) is defined similarly. Note that the intricate names of contexts, as in $\text{Ty } x:*,y:*,\alpha:x=y$ indicate their definitions which have been hidden. For instance we are assuming the definition: $x:*,y:*,\alpha:x=y = \epsilon , * , \text{var } (vS \ v0) =h \text{var } v0$

```

sym*-Ty : Ty x:*,y:*,\alpha:x=y
sym*-Ty = vY =h vX

```

```

sym*-Tm : Tm {x:*,y:*,\alpha:x=y} sym*-Ty
sym*-Tm = Coh-Contr (ext c* v0)

```

```

sym-Tm : {Γ : Con}(A : Ty Γ) → Tm (rpl-T A sym*-Ty)
sym-Tm A = rpl-tm A sym*-Tm

```

Transitivity (composition). Note that each of these cells is defined by a different choice of the contractible context Δ .

```

trans*-Ty : Ty x:*,y:*,\alpha:x=y,z:*,\beta:y=z
trans*-Ty = (vX +tm _ +tm _) =h vZ

```

```

trans*-Tm : Tm trans*-Ty
trans*-Tm = Coh-Contr (ext (ext c* v0) (vS v0))

```

```

trans-Tm : {Γ : Con}(A : Ty Γ) → Tm (rpl-T A trans*-Ty)
trans-Tm A = rpl-tm A trans*-Tm

```

For each of reflexivity, symmetry and transitivity we can construct appropriate coherence 2-cells witnessing the groupoid axioms. The base case for variable contexts is proved simply using contractibility. We use substitution to define the application of the three basic terms we have defined above.

```

Tm-right-identity* : Tm {x:*,y:*,\alpha:x=y}
  (trans*-Tm [ IdCm , vY , reflY ]tm =h v\alpha)
Tm-right-identity* = Coh-Contr (ext c* v0)

```

```

Tm-left-identity* : Tm {x:*,y:*,\alpha:x=y}
  (trans*-Tm [ ((IdCm @ pr1 @ pr1) , vX) ,
    reflX , vY , v\alpha ]tm =h v\alpha)
Tm-left-identity* = Coh-Contr (ext c* v0)

```

```

Tm-right-inverse* : Tm {x:*,y:*,\alpha:x=y}
  (trans*-Tm [ (IdCm , vX) , sym*-Tm ]tm =h reflX)
Tm-right-inverse* = Coh-Contr (ext c* v0)

```

```

Tm-left-inverse* : Tm {x:*,y:*,\alpha:x=y}
  (trans*-Tm [ ((• , vY) , vX , sym*-Tm , vY) , v\alpha ]tm =h reflY)
Tm-left-inverse* = Coh-Contr (ext c* v0)

```

```

Tm-G-assoc* : Tm Ty-G-assoc*
Tm-G-assoc* = Coh-Contr (ext (ext (ext c* v0) (vS v0)) (vS v0))

```

Their general versions are defined using replacement. For instance, for associativity, we define:

```

Tm-G-assoc : {Γ : Con}(A : Ty Γ)
  → Tm (rpl-T A Ty-G-assoc*)
Tm-G-assoc A = rpl-tm A Tm-G-assoc*

```

3.3 Higher Structure

In this section we propose how also higher groupoid structure can be introduced in the syntactical framework. We use the more abstract language of category theory to com-

municate the gist of the construction leaving the tedious formalisation for future work. To this end note that contexts and context morphisms form a category up to definitional quality. Because equality of contexts is decidable we may assume UIP on context morphisms and we are therefore working in a honest 1-category where equality of arrows is definitional equality of context morphisms. This category will be denoted **Con**.

3.3.1 Identities

For each type of level $n \in \mathbb{N}$, we have defined in Section 3.1.2 a context called *span* which has $2n+1$ variables which except for the top level, n , there are two variables on each level whose type is the equality type of the two variables on the level below, except for the bottom-level variables which are of type $*$. We call denote a span of any type of level n , S_n . Note that all such spans are isomorphic.

In each case we call the last variable the *peak*. Note that each S_n is contractible because it is a suspension of a contractible context. We call the proof of contractibility of S_n **is-contr** S_n .

In each S_n define the type σ_n as $x_{2n-2} =_h x_{2n-1}$. It is the type of the top variable. We are going to show that the following

$$S_0 \begin{array}{c} \xleftarrow{s_0} \\ \xrightarrow{t_0} \end{array} S_1 \begin{array}{c} \xleftarrow{s_1} \\ \xrightarrow{t_1} \end{array} S_2 \cdots S_n \begin{array}{c} \xleftarrow{s_n} \\ \xrightarrow{t_n} \end{array} S_{n+1} \cdots$$

is a *reflexive globular object* in **Con**. I.e. we define morphisms s_n, t_n, i_n between spans that it satisfy the following usual *globular identities*:

$$\begin{aligned} s_n t_{n+1} &= s_n s_{n+1} \\ t_n t_{n+1} &= t_n s_{n+1} \end{aligned} \quad (1)$$

together with

$$s_n i_n = \text{id} = t_n i_n \quad (2)$$

To this end, for each n , define context morphisms $s_n, t_n : S_{n+1} \Rightarrow S_n$ by the substitutions

$$\begin{aligned} s_n &= & x_k &\leftarrow x_k & k < 2n \\ & & x_{2n} &\leftarrow x_{2n} \\ t_n &= & x_k &\leftarrow x_k & k < 2n \\ & & x_{2n} &\leftarrow 2n+1 \end{aligned}$$

In words, s_n forgets the last two variables of S_{n+1} and t_n forgets the peak and its domain. It's easy to check that s and t indeed satisfy (1).

In order to define $i_n : S_n \Rightarrow S_{n+1}$, we must consider stalks (see Section 3.1), which are contexts, hereby denoted \bar{S}_n , which are like S_n without the last variable, together with types $\bar{\sigma}_n$ which are like σ_n but considered in the smaller context.

For each n there is a context morphism $\bar{i}_n : S_n \Rightarrow \bar{S}_{n+1}$ defined by

$$\bar{i}_n = \begin{array}{ll} x_k & \leftarrow x_k \\ x_{2n+1} & \leftarrow x_{2n} \end{array} \quad k \leq 2n$$

The substitution of $\bar{\sigma}_{n+1}$ along $\bar{i}_n, \bar{i}_n[\bar{\sigma}_{n+1}]_\top$, is equal to $x_{2n} =_h x_{2n}$ in S_n . So in order to extend \bar{i}_n to $i_n : S_n \Rightarrow S_{n+1}$ we must define a term in $\bar{i}_n[\bar{\sigma}_{n+1}]_\top$. We can readily do that by **coh**:

$$i_n = \bar{i}_n, \text{coh}(\text{IdCm } S_n) (x_{2n} =_h x_{2n}) (\text{is-contr } S_n)$$

It's easy to check that i_n satisfies (2).

For each n consider the chain

$$i_0^n \equiv S_0 \xrightarrow{i_0} S_1 \xrightarrow{i_1} S_2 \rightarrow \cdots \xrightarrow{i_{n-1}} S_n$$

The substitution $\sigma_n[i_0]_\top \cdots [i_{n-2}]_\top [i_{n-1}]_\top \equiv \sigma_n[i_0^n]_\top$ is a type, λ_n , in S_0 . We call λ_n the *n-iterated loop type* on x_0 . The term $S_0 \vdash (\text{var } x_{2n})[i_0^n]_{\text{tm}} : \lambda_n$ is the iterated identity term on x_0 .

3.3.2 Composition

For $m > n$ write s_n^m, t_n^m for $m-n$ iteration of s and t , respectively. Explicitly:

$$\begin{aligned} s_n^m &= s_{m-1} \cdots s_n & : S_m \Rightarrow S_n \\ t_n^m &= t_{m-1} \cdots t_n & : S_m \Rightarrow S_n \end{aligned}$$

For each $n \in \mathbb{N}$ define context Y_n by the pullback in:

$$\begin{array}{ccc} S_0 & \xleftarrow{t_0^n} & S_n \\ s_0^n \uparrow & & \uparrow l_n \\ S_n & \xleftarrow{r_n} & Y_n \end{array} \quad \sqcap$$

By definition of pullbacks, Y_n looks like a pair of spans S_n together with the proviso that the variable 1 of one is always equal to variable 0 of the other. I.e. Y_n has the shape of two spans pasted target-to-source at level 0. It is easy to check that this is indeed a pullback.

By the globular identities the two outer squares in the diagram below commute and by the universal property of the pullback imply a pair of mediating arrows as indicated.

$$\begin{array}{ccccc} S_0 & \xleftarrow{t_0^n} & S_n & \xleftarrow{s_n} & S_{n+1} \\ s_0^n \uparrow & & \uparrow l_n & & \uparrow l_{n+1} \\ S_n & \xleftarrow{r_n} & Y_n & & \\ s_n \uparrow & & \swarrow \langle s, s \rangle_n & & \\ S_{n+1} & \xleftarrow{r_{n+1}} & Y_{n+1} & & \end{array}$$

Similarly we obtain an arrow $Rr_n : Y_{n+1} \rightarrow Y_n$. The morphisms l_n and r_n provide projections onto the left and right span of Y_n respectively. The mediating arrows $\langle s, s \rangle_n$ and $\langle t, t \rangle_n$ provide projections out of Y_{n+1} onto the join of the sources and targets of the left and right parts respectively.

In order to define composition we define for each n a third morphism $c_n : Y_n \Rightarrow S_n$ with the property that both the

s -squares and t -squares below commute.

$$\begin{array}{ccc}
 S_n & \xleftarrow{s_n} & S_{n+1} \\
 \uparrow c_n & \xleftarrow{t_n} & \uparrow c_{n+1} \\
 Y_n & \xleftarrow{\langle s, s \rangle_n} & Y_{n+1} \\
 & \xleftarrow{\langle t, t \rangle_n} &
 \end{array} \quad (3)$$

The commutativity of (3) expresses the fact that the source of a composition is a composition of sources and the target of a composition is a composition of target.

It follows from all of this that for a context Γ and a pair of morphisms $a, b : \Gamma \Rightarrow S_n$, there is a context morphism $c\langle a, b \rangle : \Gamma \Rightarrow S_n$ from $s_0^n a$ to $t_0^n b$ which is the composition of a and b .

4. SEMANTICS

4.1 Globular Types

To interpret the syntax, we need globular types¹. Globular types are defined coinductively as follows.

```

record Glob : Set1 where
  constructor |-|
  field
    obj      : Set
    hom      : obj → obj → ∞ Glob
    open Glob public renaming (obj to |-|)

```

If all the object types are indeed sets, i.e. uniqueness of identity types holds, we call this a globular set.

As an example, we could construct the identity globular type called Idw .

```

Idω      : (A : Set) → Glob
Idω A    = A || (λ a b → # Idω (a ≡ b))

```

Note that this is usually not a globular set.

Given a globular set G , we can interpret the syntactic objects.

The record definition also require some semantic weakening and semantic substitution laws. The semantic weakening rules tell us how to deal with the weakening inside interpretation.

```

record Semantic (G : Glob) : Set1 where
  field
    [-]C      : Con → Set
    [-]T      : ∀ {Γ} → Ty Γ → [Γ]C → Glob
    [-]tm     : ∀ {Γ A} → Tm A → (γ : [Γ]C) → | [A]T γ |
    [-]cm     : ∀ {Γ Δ} → Γ ⇒ Δ → [Γ]C → [Δ]C

```

¹Even though we use the Agda `|Set|`, this isn't necessarily a set in the sense of Homotopy Type Theory.

π provides the projection of the semantic variable out of a semantic context.

$$\pi : \forall \{ \Gamma A \} (x : \mathbf{Var} A) (\gamma : [\Gamma] C) \rightarrow | [A] T \gamma |$$

Following are the computation laws for the interpretations of contexts and types.

$$\begin{aligned}
 [-]C-\beta 1 & : [\epsilon] C \equiv \top \\
 [-]C-\beta 2 & : \forall \{ \Gamma A \} \rightarrow [\Gamma , A] C \equiv \\
 & \quad \Sigma [\Gamma] C (\lambda \gamma \rightarrow | [A] T \gamma |) \\
 [-]T-\beta 1 & : \forall \{ \Gamma \} \{ \gamma : [\Gamma] C \} \rightarrow [*] T \gamma \equiv G \\
 [-]T-\beta 2 & : \forall \{ \Gamma A u v \} \{ \gamma : [\Gamma] C \} \\
 & \quad \rightarrow [u =_h v] T \gamma \equiv \\
 & \quad \quad b (\mathbf{hom} ([A] T \gamma) ([u] tm \gamma) ([v] tm \gamma))
 \end{aligned}$$

The semantic substitution properties are essential,

$$\begin{aligned}
 \mathbf{semSb-T} & : \forall \{ \Gamma \Delta \} (A : \mathbf{Ty} \Delta) (\delta : \Gamma \Rightarrow \Delta) (\gamma : [\Gamma] C) \\
 & \quad \rightarrow [A [\delta] T] T \gamma \equiv [A] T ([\delta] cm \gamma) \\
 \mathbf{semSb-tm} & : \forall \{ \Gamma \Delta \} \{ A : \mathbf{Ty} \Delta \} (a : \mathbf{Tm} A) (\delta : \Gamma \Rightarrow \Delta) \\
 & \quad (\gamma : [\Gamma] C) \rightarrow \mathbf{subst} _ | (semSb-T A \delta \gamma) ([a [\delta] tm] tm \gamma) \\
 & \quad \equiv [a] tm ([\delta] cm \gamma) \\
 \mathbf{semSb-cm} & : \forall \{ \Gamma \Delta \Theta \} (\gamma : [\Gamma] C) (\delta : \Gamma \Rightarrow \Delta) (\vartheta : \Delta \Rightarrow \Theta) \\
 & \quad \rightarrow [\vartheta \odot \delta] cm \gamma \equiv [\vartheta] cm ([\delta] cm \gamma)
 \end{aligned}$$

Since the computation laws for the interpretations of terms and context morphisms are well typed up to these properties.

$$\begin{aligned}
 [-]tm-\beta 1 & : \forall \{ \Gamma A \} \{ x : \mathbf{Var} A \} \{ \gamma : [\Gamma] C \} \\
 & \quad \rightarrow [\mathbf{var} x] tm \gamma \equiv \pi x \gamma \\
 [-]cm-\beta 1 & : \forall \{ \Gamma \} \{ \gamma : [\Gamma] C \} \rightarrow [\bullet] cm \gamma \equiv \\
 & \quad \mathbf{coerce} [-]C-\beta 1 \mathbf{tt} \\
 [-]cm-\beta 2 & : \forall \{ \Gamma \Delta \} \{ A : \mathbf{Ty} \Delta \} \{ \delta : \Gamma \Rightarrow \Delta \} \{ \gamma : [\Gamma] C \} \\
 & \quad \{ a : \mathbf{Tm} (A [\delta] T) \} \\
 & \quad \rightarrow [\delta , a] cm \gamma \equiv \mathbf{coerce} [-]C-\beta 2 (([\delta] cm \gamma) , \\
 & \quad \mathbf{subst} _ | (semSb-T A \delta \gamma) ([a] tm \gamma))
 \end{aligned}$$

The semantic weakening properties should actually be derivable since weakening is equivalent to projection substitution.

$$\begin{aligned}
 \mathbf{semWk-T} & : \forall \{ \Gamma A B \} (\gamma : [\Gamma] C) (v : | [B] T \gamma |) \\
 & \quad \rightarrow [A + T B] T (\mathbf{coerce} [-]C-\beta 2 (\gamma , v)) \equiv \\
 & \quad [A] T \gamma \\
 \mathbf{semWk-cm} & : \forall \{ \Gamma \Delta B \} (\gamma : [\Gamma] C) \{ v : | [B] T \gamma | \} \\
 & \quad (\delta : \Gamma \Rightarrow \Delta) \rightarrow [\delta + S B] cm \\
 & \quad (\mathbf{coerce} [-]C-\beta 2 (\gamma , v)) \equiv [\delta] cm \gamma
 \end{aligned}$$

$$\begin{aligned}
\text{semWk-tm} &: \forall \{ \Gamma \ A \ B \} (\gamma : \llbracket \Gamma \rrbracket C) (v : \llbracket B \rrbracket T \gamma) \\
(a : \text{tm } A) &\rightarrow \text{subst } _ \mid (semWk-T \gamma v) \\
(\llbracket a + \text{tm } B \rrbracket \text{tm } (\text{coerce } \llbracket _ \rrbracket C-\beta 2 (\gamma, v))) \\
&\equiv (\llbracket a \rrbracket \text{tm } \gamma)
\end{aligned}$$

Here we declare them as properties because they are essential for the computation laws of function π .

$$\begin{aligned}
\pi\text{-}\beta 1 &: \forall \{ \Gamma \ A \} (\gamma : \llbracket \Gamma \rrbracket C) (v : \llbracket A \rrbracket T \gamma) \\
&\rightarrow \text{subst } _ \mid (semWk-T \gamma v) \\
&\quad (\pi \text{v0 } (\text{coerce } \llbracket _ \rrbracket C-\beta 2 (\gamma, v))) \equiv v \\
\pi\text{-}\beta 2 &: \forall \{ \Gamma \ A \ B \} (x : \text{Var } A) (\gamma : \llbracket \Gamma \rrbracket C) (v : \llbracket B \rrbracket T \gamma) \\
&\rightarrow \text{subst } _ \mid (semWk-T \gamma v) (\pi \text{vS } \{ \Gamma \} \{ A \} \{ B \} x) \\
&\quad (\text{coerce } \llbracket _ \rrbracket C-\beta 2 (\gamma, v))) \\
&\equiv \pi x \gamma
\end{aligned}$$

The only part of the semantics where we have any freedom is the interpretation of the coherence constants:

$$\begin{aligned}
\llbracket \text{coh} \rrbracket &: \forall \{ \Theta \} \rightarrow \text{isContr } \Theta \rightarrow (A : \text{Ty } \Theta) \\
&\rightarrow (\vartheta : \llbracket \Theta \rrbracket C) \rightarrow \llbracket A \rrbracket T \vartheta
\end{aligned}$$

However, we also need to require that the coherence constants are well behaved wrt to substitution which in turn relies on the interpretation of all terms. To address this we state the required properties in a redundant form because the correctness for any other part of the syntax follows from the defining equations we have already stated. However, there seems to be no way to avoid this.

If the underlying globular type is not a globular set we need to add coherence laws, which is not very well understood. On the other hand, restricting ourselves to globular sets means that our prime example $\text{Id}\omega$ is not an instance anymore. We should still be able to construct non-trivial globular sets, e.g. by encoding basic topological notions and defining higher homotopies as in a classical framework. However, we don't currently know a simple definition of a globular set which is a weak ω -groupoid. One possibility would be to use the syntax of type theory with equality types. Indeed, we believe that this would be an alternative way to formalize weak ω -groupoids.

5. CONCLUSION

In this paper, we present an implementation of weak ω -groupoids following Brunerie's work. Briefly speaking, we define the syntax of the type theory $\mathcal{T}_{\infty\text{-groupoid}}$, then a weak ω -groupoid is a globular set with the interpretation of the syntax. To overcome some technical problems, we use heterogeneous equality for terms, some auxiliary functions and loop context in all implementation. We construct the identity morphisms and verify some groupoid laws in the syntactic framework. The suspensions for all sorts of objects are also defined for other later constructions.

There is still a lot of work to do within the syntactic framework. For instance, we would like to investigate the relation between the $\mathcal{T}_{\infty\text{-groupoid}}$ and a Type Theory with equality types and J eliminator which is called \mathcal{T}_{eq} . One direction is to simulate the J eliminator syntactically in $\mathcal{T}_{\infty\text{-groupoid}}$ as we mentioned before, the other direction is to derive J using *coh* if we can prove that the \mathcal{T}_{eq} is a weak ω -groupoid. The syntax could be simplified by adopting categories with families. An alternative may be to use higher inductive types directly to formalize the syntax of type theory.

We would like to formalise a proof of that $\text{Id}\omega$ is a weak ω -groupoid, but the base set in a globular set is an h-set which is incompatible with $\text{Id}\omega$. Perhaps we could solve the problem by instead proving a syntactic result, namely that the theory we have presented here and the theory of equality types with J are equivalence. Finally, to model the Type Theory with weak ω -groupoids and to eliminate the univalence axiom would be the most challenging task in the future.

6. REFERENCES

- [1] Thorsten Altenkirch. Extensional Equality in Intensional Type Theory. In *14th Symposium on Logic in Computer Science*, pages 412 – 420, 1999.
- [2] Thorsten Altenkirch, Conor McBride, and Wouter Swierstra. Observational equality, now! In *PLPV '07: Proceedings of the 2007 workshop on Programming languages meets program verification*, pages 57–68, New York, NY, USA, 2007. ACM.
- [3] Thorsten Altenkirch and Ondrej Rypacek. A syntactical Approach to Weak ω -groupoids. In *Computer Science Logic (CSL'12) - 26th International Workshop/21st Annual Conference of the EACSL, CSL 2012*, 2012.
- [4] D. Ara. On the homotopy theory of Grothendieck ∞ -groupoids. *ArXiv e-prints*, June 2012.
- [5] Andrej Bauer and Peter LeFanu Lumsdaine. A Coq proof that Univalence Axioms implies Functional Extensionality. 2013.
- [6] Guillaume Brunerie. Syntactic Grothendieck weak ∞ -groupoids. <http://uf-ias-2012.wikispaces.com/file/view/SyntacticInfinityGroupoidsRawDefinition.pdf>, 2013.
- [7] Alexander Grothendieck. Pursuing Stacks. 1983. Manuscript.
- [8] G. Maltsiniotis. Grothendieck ∞ -groupoids, and still another definition of ∞ -categories. *ArXiv e-prints*, September 2010.
- [9] The Univalent Foundations Program. Homotopy type theory: Univalent foundations of mathematics. Technical report, Institute for Advanced Study, 2013.