Some constructions on ω-groupoids

Thorsten Altenkirch
University of Nottingham
United Kingdom
txa@cs.nott.ac.uk

Nuo Li
University of Nottingham
United Kingdom
nzl@cs.nott.ac.uk

Ondřej Rypáček
University of Oxford
United Kingdom
ondrej.rypacek@gmail.com

ABSTRACT

Weak ω-groupoids are the higher dimensional generalisation of setoids and are an essential ingredient of the constructive semantics of Homotopy Type Theory [9]. Following up on our previous formalisation [3] and Brunerie's notes [5], we present a new formalisation of the syntax of weak ω-groupoids in Agda using heterogeneous equality. We show how to recover basic constructions on ω-groupoids using suspension and replacement. In particular we show that any type forms a groupoid and that we can derive higher dimensional composition. We present a possible semantics using globular sets and discuss the issues which arise when using globular types instead.

Categories and Subject Descriptors

F.4.1 [**Mathematical Logic and Formal Languages**]: Lambda calculus and related systems, Mechanical theorem proving

General Terms

Theory

Keywords

Type Theory, Homotopy Type Theory, Category Theory, Higher dimensional structures, Agda

1. INTRODUCTION

In Type Theory, a type can be interpreted as a setoid which is a set equipped with an equivalence relation [1]. The equivalence proof of the relation consists of reflexivity, symmetry and transitivity whose proofs are unique. However in Homotopy Type Theory, we reject the principle of uniqueness of identity proofs (UIP). Instead we accept the univalence axiom which says that equality of types is weakly equivalent to weak equivalence. Weak equivalence can be seen as a refinement of isomorphism without UIP [3]. For example, a weak equivalence between two objects A and B in a 2-category is a morphism $f : A \longrightarrow B$ which has a corresponding inverse morphism $g : B \longrightarrow A$, but instead of the proofs of isomorphism $f \circ g = 1_B$ and $g \circ f = 1_A$ we have two 2-cell isomorphisms $f \circ g \cong 1_B$ and $g \circ f \cong 1_A$.

Voevodsky proposed the univalence axiom which basically says that isomorphic types are equal. This can be viewed as a strong extensionality axiom and it does imply functional extensionality. However, adding univalence as an axiom destroys canonicity, i.e. that every closed term of type $\mathbb{N}$ is reducible to a numeral. In the special case of extensionality and assuming a strong version of UIP we were able to eliminate this issue [1, 2] using setoids. However, it is not clear how to generalize this in the absence of UIP to univalence which is incompatible with UIP. To solve the problem we should generalise the notion of setoids, namely to enrich the structure of the identity proofs.

The generalised notion is called weak ω-groupoids and was proposed by Grothendieck 1983 in a famous manuscript *Pursuing Stacks* [6]. Maltsiniotis continued his work and suggested a simplification of the original definition which can be found in [8]. Later Ara also presents a slight variation of the simplication of weak ω-groupoids in [4]. Categorically speaking an ω-groupoid is an ω-category in which morphisms on all levels are equivalences. As we know that a set can be seen as a discrete category, a setoid is a category where every morphism is unique between two objects. A groupoid is more generalised, every morphism is isomorphism but the proof of isomorphism is unique, namely the composition of a morphism with its inverse is equal to an identity morphism. Similarly, an n-groupoid is an n-category in which morphisms on all levels are equivalence. ω-groupoids which are also called ∞-groupoids is an infinite version of n-groupoids. To model Type Theory without UIP we also require the equalities to be non-strict, in other words, they are not definitionally equalities. Finally we should use weak ω-groupoids to interpret types and eliminate the univalence axiom.

There are several approaches to formalise weak ω-groupoids in Type Theory. For instance, Altenkirch and Rypáček [3], and Brunerie's notes [5]. This paper mainly explains an implementation of weak ω-groupoids following Brunerie's approach in Agda which is a well-known theorem prover and also a variant of intensional Martin-Löf type theory. This is the first attempt to formalise this approach in dependently typed languages like Agda and Coq. The approach is to specify when a globular set is a weak ω-groupoid by first defining a type theory called $\mathcal{T}_{\infty-groupoid}$ to describe the internal language of Grothendieck weak ω-groupoids, then interpret it with a globular set and a dependent function. All coherence laws of the weak ω-groupoids should be derivable from the syntax, we will present some basic ones, for example reflexivity. One of the main contributions of this paper is to use the heterogeneous equality for terms to overcome some very difficult problems when we used the normal homogeneous one. In this paper, we omit some complicated and less important programs, namely the proofs of some lemmas or the definitions of some auxiliary functions. It is still possible for the reader who is interested in the details to check the code online, in which there are only some minor differences.

1.1 Agda

1. Agda is a dependently typed programming languages which offers a set of tools for theorem proving. It is a variant of Martin-Löf type theory. A good introductory resources can be found on Agda wiki.

2. short introduction to Agda syntax

Acknowledgements

The first and second author would like to thank the organizers and other participants of the special year on homotopy type theory at the Institute for Advanced Study where they had many interesting discussion topics related to the work presented in this paper. Especially we all would like to acknowledge Guillaume Brunerie's proposal which made this work possible.

2. SYNTAX

We develop the type theory of ω-groupoids formally, following [5]. This is a Type Theory with only one type former which we can view as equality types and interpret as the homsets of the ω-groupoid. There are no definitional equalities which correspond to the fact that we consider weak ω-groupoids. None of the groupoid laws on any levels are strict (i.e. definitional) but all are witnessed by terms. Compared to [3] the definition is very much simplified by the observation that all laws of a weak ω-groupoid follow from the existence of coherence constants for any contractible context.

In our formalisation we exploit the more liberal way to do mutual definitions in Agda, which was implemented recently following up a suggestion by the first author. It allows us to first introduce a type former but give its definition later.

Since we are avoiding definitional equalities we have to define a syntactic substitution operation which we need for the general statement of the coherence constants. However, defining this constant requires us to prove a number of substitution laws at the same time. We address this issue by using a heterogeneous equality which exploits UIP. Note that UIP holds for the syntax because all components defined here are sets in the sense of Homotopy Type Theory.

2.1 Basic Objects

We first declare the syntax of our type theory which is called $\mathcal{T}_{\infty-groupoid}$ namely the internal language of weak ω-groupoids. The following declarations in order are contexts as sets, types are sets dependent on contexts, terms and variables are sets dependent on types, Contexts morphisms and the contractible contexts. Since the definitions of them involve each other, it is necessary to define them inductive-inductively. Therefore we claim their types first and then define the constructors later.

```
data Con           : Set
data Ty (Γ : Con)  : Set
data Tm            : {Γ : Con}(A : Ty Γ) → Set
data Var           : {Γ : Con}(A : Ty Γ) → Set
data _⇒_           : Con → Con → Set
data isContr       : Con → Set
```

Contexts are inductively defined as either an empty context or a context with a type in it.

```
data Con where
  ε   : Con
  _,_ : (Γ : Con)(A : Ty Γ) → Con
```

Types are defined as either $*$ which we call 0-cells, or a equality type between two terms of some type A. If type A is n-cell then we call its equality type $(n+1)$-cell.

```
data Ty Γ where
  *     : Ty Γ
  _=h_  : {A : Ty Γ}(a b : Tm A) → Ty Γ
```

2.2 Heterogeneous Equality for Terms

One of the big challenges we encountered at first is the difficulty to formalise and to reason about the equalities of terms, which is essential when defining substitution. When the usual identity types are used which are homogeneous, one has to use substitution to unify the types on both sides of equality types. This results in *subst* to appear in terms, about which one has to state substitution lemmas. This further pollutes syntax requiring lemmas about lemmas, lemmas about lemmas about lemmas, etc. The resulting recurrence pattern has been identified and implemented in [3] for the special cases of coherence cells for associativity, units and interchange. However it is not clear how that approach could be adapted to the present, much more economical formulation of weak ω-groupoids. Moreover, the complexity brings the Agda type checker to its limits and correctness into question.

For example, assume we have a context Γ, types $A\ B : Ty\ \Gamma$ and a term of the equality $p : A \equiv B$. If we have a term $a : Tm\ A$ and a term $b : Tm\ B$, it is impossible to just write $a \equiv b$ because of type unification. We need to write $subst\ Tm\ pa \equiv b$. Again assume we have another type $C : Ty\ \Gamma$, $q : B \equiv C$ and a term of $c : Tm\ C$, to prove a possible lemma $a = c$, we have to write $subst\ Tm\ q\ (subst\ Tm\ p\ a) \equiv c$. Of course now we need to prove a new coherence lemma $subst\ Tm\ q\ (subst\ Tm\ p\ a) \equiv subst\ Tm\ (trans\ p\ q)\ a$ to help us in other proofs, etc.

The idea of heterogenous equality, which we use to resolve this issue, is that one can define equality for terms of different types, but its inhabitants only for terms of definitionally equal types. However, the corresponding elimination principle relies on UIP.

In intensional type theory, UIP is not provable in general, namely not all types are h-sets (homotopy 0-types). However it is justified to claim all type with decidable equality are h-sets. From the Hedberg's Theorem [7] we know that inductive types with finitary constructors have decidable equality. In our case, the types which stand for syntactic components (contexts, types, terms) are inductive-inductive types with finitary constructors and it is therefore safe to assume UIP holds for them. We can therefore safely use heterogenous equality for the syntax because its equality, which will be introduced later, is decidable.

In summary, the equality of syntactic types is unique, so it is safe to use heterogeneous equality and do without the substitution lemmas which would otherwise be necessary to match terms of different types.

```
data _≅_ {Γ : Con}{A : Ty Γ} :
      {B : Ty Γ} → Tm A → Tm B → Set where
  refl : (b : Tm A) → b ≅ b
```

Once we have the heterogeneous equality for terms, we can define a proof-irrelevant substitution which we call coercion here since it gives us a term of type A if we have a term of type B and the two types are equal. We can also prove that the coerced term is heterogeneously equal to the original term. Combining these definitions, it is much more convenient to formalise and to reason about term equations.

```
_⟦_⟫         : {Γ : Con}{A B : Ty Γ}(a : Tm B)
               → A ≡ B → Tm A
a ⟦ refl ⟫    = a

cohOp        : {Γ : Con}{A B : Ty Γ}{a : Tm B}(p : A ≡ B)
               → a ⟦ p ⟫ ≅ a
cohOp refl   = refl _
```

2.3 Substitutions

With context morphisms, we can define substitutions for types variables and terms. Usually we define a set of symbols together and we name a function $*$ as $*C$ for contexts, $*T$ for types, $*V$ for variables $*tm$ for terms and $*S$ (or $*cm$) for context morphisms. For example the substitution for types is defined as follows

```
_[_]T    : ∀{Γ Δ} → Ty Δ → Γ ⇒ Δ → Ty Γ
_[_]V    : ∀{Γ Δ A} → Var A → (δ : Γ ⇒ Δ) → Tm (A [ δ ]T)
_[_]tm   : ∀{Γ Δ A} → Tm A → (δ : Γ ⇒ Δ) → Tm (A [ δ ]T)
```

Indeed the composition of contexts can be understood as substitution for context morphisms as well.

```
_⊚_ : ∀{Γ Δ Θ} → Δ ⇒ Θ → (δ : Γ ⇒ Δ) → Γ ⇒ Θ
```

2.4 Weakening Rules

We can freely add types to the contexts of any given type judgments, term judgments or context morphisms. These are weakening rules.

```
_+T_   : ∀{Γ}(A : Ty Γ)(B : Ty Γ) → Ty (Γ , B)
_+tm_  : ∀{Γ}{A : Ty Γ}(a : Tm A)(B : Ty Γ) → Tm (A +T B)
_+S_   : ∀{Γ Δ}(δ : Γ ⇒ Δ)(B : Ty Γ) → (Γ , B) ⇒ Δ
```

To define variables we have to use the weakening rules. We use typed de Bruijn indices to define variables as either the rightmost variable of the context, or some variable in the context which can be found by cancelling the rightmost variable along with each vS. The coherence constants are one of the major part of this syntax, which are primitive terms of the primitive types in contractible contexts which will be introduced later. Since contexts, types, variables and terms are all mutually defined, most of their properties have to be proved simultaneously.

```
data Var where
  v0 : {Γ : Con}{A : Ty Γ}     → Var (A +T A)
  vS : {Γ : Con}{A B : Ty Γ}(x : Var A) → Var (A +T B)
```

A term can be either a variable or a coherence constant (coh). It encodes all constants for arbitrary types in a contractible context.

```
data Tm where
  var : {Γ : Con}{A : Ty Γ} → Var A → Tm A
  coh : {Γ Δ : Con} → isContr Δ → (δ : Γ ⇒ Δ)
      → (A : Ty Δ) → Tm (A [ δ ]T)
```

With variables defined, it is possible to formalise another core part of the syntactic framework, *contractible contexts*. Intuitively speaking, a context is contractible if its geometric realization is contractible to a point. It either contains one variable of the 0-cell $*$ which is the base case, or we can extend a contractible context with a variable of an existing type and an n-cell, namely a morphism, between the new variable and some existing variable. The graph can be drawn like branching trees.

```
data isContr where
  c*   : isContr (ε , *)
  ext  : ∀{Γ} → isContr Γ → {A : Ty Γ}(x : Var A)
       → isContr (Γ , A , (var (vS x) =h var v0))
```

Context morphisms are defined inductively similarly to contexts. A context morphism is a list of terms corresponding to the list of types in the context on the right hand side of the morphism.

```
data _⇒_ where
  •     : ∀{Γ} → Γ ⇒ ε
  _,_   : ∀{Γ Δ}(δ : Γ ⇒ Δ){A : Ty Δ}
    (a : Tm (A [ δ ]T)) → Γ ⇒ (Δ , A)
```

2.5 Lemmas

The following lemmas are essential for constructions and theorem proving later. The first set of lemmas states that to substitute a type, a variable, a term, or a context morphism with two context morphisms consecutively, is equivalent to substitute with the composition of the two context morphisms:

```
[⊚]T    : ∀{Γ Δ Θ}{A : Ty Θ}{ϑ : Δ ⇒ Θ}
           {δ : Γ ⇒ Δ} → A [ ϑ ⊚ δ ]T ≡ (A [ ϑ ]T)[ δ ]T
```

The second set states that weakening inside substitution is equivalent to weakening outside:

```
[+S]T    : ∀{Γ Δ}{A : Ty Δ}{δ : Γ ⇒ Δ}{B : Ty Γ}
           → A [ δ +S B ]T ≡ (A [ δ ]T) +T B
```

$$[+S]\mathsf{tm} \quad : \forall\{\Gamma\ \Delta\}\{A : \mathsf{Ty}\ \Delta\}(a : \mathsf{Tm}\ A)\{\delta : \Gamma \Rightarrow \Delta\}\{B : \mathsf{Ty}\ \Gamma\} \to a\ [\ \delta\ {+S}\ B\ ]\mathsf{tm} \cong (a\ [\ \delta\ ]\mathsf{tm})\ {+\mathsf{tm}}\ B$$

$$[+S]S \quad : \forall\{\Gamma\ \Delta\ \Theta\}\{\delta : \Delta \Rightarrow \Theta\}\{\gamma : \Gamma \Rightarrow \Delta\}\{B : \mathsf{Ty}\ \Gamma\} \to \delta \circledcirc (\gamma\ {+S}\ B) \equiv (\delta \circledcirc \gamma)\ {+S}\ B$$

We can cancel the last term in the substitution for weakened objects since weakening doesn't introduce new variables in types and terms.

$$+\mathsf{T}[,]\mathsf{T} \quad : \forall\{\Gamma\ \Delta\}\{A : \mathsf{Ty}\ \Delta\}\{\delta : \Gamma \Rightarrow \Delta\}\{B : \mathsf{Ty}\ \Delta\}\{b : \mathsf{Tm}\ (B\ [\ \delta\ ]\mathsf{T})\} \to (A\ {+\mathsf{T}}\ B)\ [\ \delta\ ,\ b\ ]\mathsf{T} \equiv A\ [\ \delta\ ]\mathsf{T}$$

$$+\mathsf{tm}[,]\mathsf{tm} \quad : \forall\{\Gamma\ \Delta\}\{A : \mathsf{Ty}\ \Delta\}(a : \mathsf{Tm}\ A)\{\delta : \Gamma \Rightarrow \Delta\}\{B : \mathsf{Ty}\ \Delta\}\{c : \mathsf{Tm}\ (B\ [\ \delta\ ]\mathsf{T})\} \to (a\ {+\mathsf{tm}}\ B)\ [\ \delta\ ,\ c\ ]\mathsf{tm} \cong a\ [\ \delta\ ]\mathsf{tm}$$

Most of the substitutions are defined as usual, except the one for coherence constants. We do substitution in the context morphism part of the coherence constants.

$$\begin{array}{ll} \mathsf{var}\ x & [\ \delta\ ]\mathsf{tm} = x\ [\ \delta\ ]\mathsf{V} \\ \mathsf{coh}\ c\Delta\ \gamma\ A & [\ \delta\ ]\mathsf{tm} = \mathsf{coh}\ c\Delta\ (\gamma \circledcirc \delta)\ A\ [\![\ \mathsf{sym}\ [\circledcirc]\mathsf{T}\ \rangle\!\rangle \end{array}$$

3. SOME IMPORTANT DERIVABLE CONSTRUCTIONS

In this section we show that it is possible to reconstruct the structure of a (weak) ω-groupoid from the syntactical framework presented in Section 2 in the style of [3]. To this end, let us call a term $a : \mathsf{Tm}\ A$ an n-cell if $\mathsf{level}\ A \equiv n$, where

$$\begin{array}{ll} \mathsf{level} & : \forall\ \{\Gamma\} \to \mathsf{Ty}\ \Gamma \to \mathbb{N} \\ \mathsf{level}\ * & = 0 \\ \mathsf{level}\ (\_{=_h}\_\ \{A\}\ \_\ \_) & = \mathsf{suc}\ (\mathsf{level}\ A) \end{array}$$

In any ω-category, any n-cell a has a domain (source), $s^n_m\, a$, and a codomain (target), $s^n_m\, a$, for each $m \leq n$. These are, of course, (n-m)-cells. For each pair of n-cells such that for some m $s^n_m a \equiv t^n_m b$, there must exist their composition $a \circ^n_m b$ which is an n-cell. Composition is (weakly) associative. Moreover for any (n-m)-cell x there exists an n-cell $\mathsf{id}^n_m\, x$ which behaves like a (weak) identity with respect to $\circ^n_m$. For the time being we discuss only the construction of cells and omit the question of coherence.

For instance, in the simple case of bicategories, each 2-cell a has a horizontal source $s^1_1\, a$ and target $t^1_1\, a$, and also a vertical source $s^2_1\, a$ and target $t^2_1 a$, which is also the source and target, of the horizontal source and target, respectively, of a.

There is horizontal composition of 1-cells $\circ^1_1$: $x \xrightarrow{f} y \xrightarrow{g} z$, and also horizontal composition of 2-cells $\circ^2_1$, and vertical composition of 2-cells $\circ^2_2$. There is a horizontal identity on a, $\mathsf{id}^1_1\, a$, and vertical identity on a, $\mathsf{id}^2_1\, a = \mathsf{id}^2_2 \mathsf{id}^1_1\, a$.

Thus each ω-groupoid construction is defined with respect to a *level*, m, and depth n-m and the structure of an ω-groupoid is repeated on each level. As we are working purely syntactically we may make use of this fact and define all groupoid structure only at level $m = 1$ and provide a so-called *replacement operation* which allows us to lift any cell to an arbitrary type A. It is called 'replacement' because we are syntactically replacing the base type $*$ with an arbitrary type, A.

An important general mechanism we rely on throughout the development follows directly from the type of the only non-trivial constructor of Tm, coh, which tells us that to construct a new term of type $\Gamma \vdash A$, we need a contractible context, Δ, a type $\Delta \vdash T$ and a context morphism $\delta : \Gamma \Rightarrow \Delta$ such that

$$T[\,\delta\,]\mathsf{T} \equiv A$$

Because in a contractible context all types are inhabited we may in a way work freely in Δ and then pull back all terms to A using δ. To show this formally, we must first define identity context morphisms which complete the definition of a *category* of contexts and context morphisms:

$$\mathsf{IdCm} : \forall\ \{\Gamma\} \to \Gamma \Rightarrow \Gamma$$

It satisfies the following property:

$$\mathsf{IC\text{-}T} \quad : \forall\{\Gamma : \mathsf{Con}\}\{A : \mathsf{Ty}\ \Gamma\} \to A\ [\ \mathsf{IdCm}\ ]\mathsf{T} \equiv A$$

The definition proceeds by structural recursion and therefore extends to terms, variables and context morphisms with analogous properties. It allows us to define at once:

$$\begin{array}{ll} \mathsf{Coh\text{-}Contr} & : \{\Gamma : \mathsf{Con}\}\{A : \mathsf{Ty}\ \Gamma\} \to \mathsf{isContr}\ \Gamma \to \mathsf{Tm}\ A \\ \mathsf{Coh\text{-}Contr}\ isC & = \mathsf{coh}\ isC\ \mathsf{IdCm}\ \_\ [\![\ \mathsf{sym}\ \mathsf{IC\text{-}T}\ \rangle\!\rangle \end{array}$$

We use Coh-Contr as follows: for each kind of cell we want to define, we construct a minimal contractible context built out of variables together with a context morphism that populates the context with terms and a lemma that states a definitional equality between the substitution and the original type.

3.1 Suspension and Replacement

For an arbitrary type A in Γ of level n one can define a context with $2n$ variables, called the *stalk* of A. Moreover one can define a morphism from Γ to the stalk of A such that its substitution into the maximal type in the stalk of A gives back A. The stalk of A depends only on the level of A, the terms in A define the substitution. Here is an example of stalks of small levels: ε (the empty context) for $n = 0$; $(x_0 : *, x_1 : *)$ for $n = 1$; $(x_0 : *, x_1 : *, x_2 : x_0 =_{\mathsf{h}} x_1, x_3 : x_0 =_{\mathsf{h}} x_1)$ for $n = 2$, etc.

This is the $\Delta = \varepsilon$ case of a more general construction where in we *suspend* an arbitrary context Δ by adding $2n$ variables to the beginning of it, and weakening the rest of the variables appropriately so that type $*$ becomes $x_{2n-2} =_{\mathsf{h}} x_{2n-1}$. A crucial property of suspension is that it preserves contractibility.

3.1.1 Suspension

Suspension is defined by iteration level-A-times the following operation of one-level suspension. $\Sigma\mathsf{C}$ takes a context and gives a context with two new variables of type $*$ added at the beginning, and with all remaining types in the context suspended by one level.

$$\begin{array}{l} \Sigma\mathsf{C} : \mathsf{Con} \to \mathsf{Con} \\ \Sigma\mathsf{T} : \{\Gamma : \mathsf{Con}\} \to \mathsf{Ty}\ \Gamma \to \mathsf{Ty}\ (\Sigma\mathsf{C}\ \Gamma) \end{array}$$

$$
\begin{aligned}
&\mathsf{\Sigma C}\ \varepsilon && = \varepsilon\ ,\ *\ ,\ * \\
&\mathsf{\Sigma C}\ (\Gamma\ ,\ A) && = \mathsf{\Sigma C}\ \Gamma\ ,\ \mathsf{\Sigma T}\ A
\end{aligned}
$$

The rest of the definitions is straightforward by structural recursion. In particular we suspend variables, terms and context morphisms:

$$
\begin{aligned}
&\mathsf{\Sigma v} && : \{\Gamma : \mathsf{Con}\}\{A : \mathsf{Ty}\ \Gamma\} \to \mathsf{Var}\ A \to \mathsf{Var}\ (\mathsf{\Sigma T}\ A) \\
&\mathsf{\Sigma tm} && : \{\Gamma : \mathsf{Con}\}\{A : \mathsf{Ty}\ \Gamma\} \to \mathsf{Tm}\ A \to \mathsf{Tm}\ (\mathsf{\Sigma T}\ A) \\
&\mathsf{\Sigma s} && : \{\Gamma\ \Delta : \mathsf{Con}\} \to \Gamma \Rightarrow \Delta \to \mathsf{\Sigma C}\ \Gamma \Rightarrow \mathsf{\Sigma C}\ \Delta
\end{aligned}
$$

The following lemma establishes preservation of contractibility by one-step suspension:

$$
\mathsf{\Sigma C\text{-}Contr} : (\Delta : \mathsf{Con}) \to \mathsf{isContr}\ \Delta \to \mathsf{isContr}\ (\mathsf{\Sigma C}\ \Delta)
$$

It is also essential that suspension respects weakening and substitution:

$$
\begin{aligned}
&\mathsf{\Sigma T[+T]} && : \{\Gamma : \mathsf{Con}\}(A : \mathsf{Ty}\ \Gamma)(B : \mathsf{Ty}\ \Gamma) \\
& && \to \mathsf{\Sigma T}\ (A \mathbin{+\mathsf{T}} B) \equiv \mathsf{\Sigma T}\ A \mathbin{+\mathsf{T}} \mathsf{\Sigma T}\ B \\
&\mathsf{\Sigma tm[+tm]} && : \{\Gamma : \mathsf{Con}\}\{A : \mathsf{Ty}\ \Gamma\}(a : \mathsf{Tm}\ A)(B : \mathsf{Ty}\ \Gamma) \\
& && \to \mathsf{\Sigma tm}\ (a \mathbin{+\mathsf{tm}} B) \cong \mathsf{\Sigma tm}\ a \mathbin{+\mathsf{tm}} \mathsf{\Sigma T}\ B \\
&\mathsf{\Sigma T[\Sigma s]T} && : \{\Gamma\ \Delta : \mathsf{Con}\}(A : \mathsf{Ty}\ \Delta)(\delta : \Gamma \Rightarrow \Delta) \\
& && \to (\mathsf{\Sigma T}\ A)\ [\ \mathsf{\Sigma s}\ \delta\ ]\mathsf{T} \equiv \mathsf{\Sigma T}\ (A\ [\ \delta\ ]\mathsf{T})
\end{aligned}
$$

General suspension to the level of a type A is defined by iteration of one-level suspension. For symmetry and ease of reading the following suspension functions take as a parameter a type A in Γ, while they depend only on its level.

$$
\mathsf{\Sigma C\text{-}it} : \{\Gamma : \mathsf{Con}\}(A : \mathsf{Ty}\ \Gamma) \to \mathsf{Con} \to \mathsf{Con}
$$

$$
\mathsf{\Sigma T\text{-}it} : \{\Gamma\ \Delta : \mathsf{Con}\}(A : \mathsf{Ty}\ \Gamma) \to \mathsf{Ty}\ \Delta \to \mathsf{Ty}\ (\mathsf{\Sigma C\text{-}it}\ A\ \Delta)
$$

$$
\begin{aligned}
&\mathsf{\Sigma tm\text{-}it} : \{\Gamma\ \Delta : \mathsf{Con}\}(A : \mathsf{Ty}\ \Gamma)\{B : \mathsf{Ty}\ \Delta\} \\
&\quad \to \mathsf{Tm}\ B \to \mathsf{Tm}\ (\mathsf{\Sigma T\text{-}it}\ A\ B)
\end{aligned}
$$

Finally, it is clear that iterated suspension preserves contractibility.

$$
\begin{aligned}
&\mathsf{\Sigma C\text{-}it\text{-}Contr} : \forall\ \{\Gamma\ \Delta\}(A : \mathsf{Ty}\ \Gamma) \to \mathsf{isContr}\ \Delta \\
&\quad \to \mathsf{isContr}\ (\mathsf{\Sigma C\text{-}it}\ A\ \Delta)
\end{aligned}
$$

By suspending the minimal contractible context, $*$, we obtain a so-called span. They are also stalks with a top variable added. For example $(x_0 : *)$ (the one-variable context) for $n = 0$; $(x_0 : *, x_1 : *, x_2 : x_0 =_{\mathsf{h}} x_1)$ for $n = 1$; $(x_0 : *, x_1 : *, x_2 : x_0 =_{\mathsf{h}} x_1, x_3 : x_0 =_{\mathsf{h}} x_1, x_4 : x_2 =_{\mathsf{h}} x_3)$ for $n = 2$, etc. Spans play an important role later in the definition of composition.

3.1.2 Replacement

After we have suspended a context by inserting an appropriate number of variables, we may proceed to a substitution which fills the stalk for A with A. The context morphism representing this substitution is called filter. In the final step we combine it with Γ, the context of A. The new context contains two parts, the first is the same as Γ, and the second is the suspended Δ substituted by filter. However, we also have to drop the stalk of A becuse it already exists in Γ.

Geometrically speaking, the context resulting from replacing $*$ in Δ by A is a new context which corresponds to the pasting of Δ to Γ to A.

As always, we define replacement for contexts, types and terms:

$$
\begin{aligned}
&\mathsf{rpl\text{-}C} && : \{\Gamma : \mathsf{Con}\}(A : \mathsf{Ty}\ \Gamma) \to \mathsf{Con} \to \mathsf{Con} \\
&\mathsf{rpl\text{-}T} && : \{\Gamma\ \Delta : \mathsf{Con}\}(A : \mathsf{Ty}\ \Gamma) \to \mathsf{Ty}\ \Delta \to \mathsf{Ty}\ (\mathsf{rpl\text{-}C}\ A\ \Delta) \\
&\mathsf{rpl\text{-}tm} && : \{\Gamma\ \Delta : \mathsf{Con}\}(A : \mathsf{Ty}\ \Gamma)\{B : \mathsf{Ty}\ \Delta\} \to \mathsf{Tm}\ B \\
& && \to \mathsf{Tm}\ (\mathsf{rpl\text{-}T}\ A\ B)
\end{aligned}
$$

Replacement for contexts, rpl-C, defines for a type A in Γ and another context Δ a context which begins as Γ and follows by each type of Δ with $*$ replaced with (pasted onto) A. To this end we must define the substitution filter which pulls back each type from suspended Δ to the new context.

$$
\begin{aligned}
&\mathsf{filter} : \{\Gamma : \mathsf{Con}\}(\Delta : \mathsf{Con})(A : \mathsf{Ty}\ \Gamma) \\
&\quad \to \mathsf{rpl\text{-}C}\ A\ \Delta \Rightarrow \mathsf{\Sigma C\text{-}it}\ A\ \Delta
\end{aligned}
$$

$$
\begin{aligned}
&\mathsf{rpl\text{-}C}\ \{\Gamma\}\ A\ \varepsilon && = \Gamma \\
&\mathsf{rpl\text{-}C}\ A\ (\Delta\ ,\ B) && = \mathsf{rpl\text{-}C}\ A\ \Delta\ ,\ \mathsf{rpl\text{-}T}\ A\ B
\end{aligned}
$$

$$
\mathsf{rpl\text{-}T}\ A\ B = \mathsf{\Sigma T\text{-}it}\ A\ B\ [\ \mathsf{filter}\ \_\ A\ ]\mathsf{T}
$$

3.2 First-level Groupoid Structure

We can proceed to the definition of the groupoid structure of the syntax. We start with the base case: 1-cells. Replacement defined above allows us to lift this structure to an arbitrary level n (we leave most of the routine details out). This shows that the syntax is a 1-groupoid on each level. In the next section we show how also the higher-groupoid structure can be defined.

We start by an essential lemma which formalises the discussion at the beginning of this Section: to construct a term in a type A in an arbitrary context, we first restrict attention to a suitable contractible context Δ and use lifting and substitution – replacement – to pull the term built by coh in Δ back. This relies on the fact that a lifted contractible context is also contractible, and therefore any type lifted from a contractible context is also inhabited.

$$
\begin{aligned}
&\mathsf{Coh\text{-}rpl} : \{\Gamma\ \Delta : \mathsf{Con}\}(A : \mathsf{Ty}\ \Gamma)(B : \mathsf{Ty}\ \Delta) \to \mathsf{isContr}\ \Delta \\
&\quad \to \mathsf{Tm}\ \{\mathsf{rpl\text{-}C}\ A\ \Delta\}\ (\mathsf{rpl\text{-}T}\ A\ B) \\
&\mathsf{Coh\text{-}rpl}\ \{\Delta = \Delta\}\ A\ B\ isc = \\
&\quad \mathsf{coh}\ (\mathsf{\Sigma C\text{-}it\text{-}\varepsilon\text{-}Contr}\ A\ isc)\ (\mathsf{filter}\ \Delta\ A)\ (\mathsf{\Sigma T\text{-}it}\ A\ B)
\end{aligned}
$$

Next we define the reflexivity, symmetry and transitivity terms of any type. Let's start from the basic case as for the base type $*$.

Reflexivity It is trivially inhabited because the context is the basic case of a contractible context.

$$
\begin{aligned}
&\mathsf{refl*} : \mathsf{Tm}\ \{\mathsf{x{:}*}\}\ (\mathsf{var\ v0 =h\ var\ v0}) \\
&\mathsf{refl*} = \mathsf{Coh\text{-}Contr\ c*}
\end{aligned}
$$

To obtain the reflexivity term for any given type, we just use replacement.

$$
\begin{aligned}
&\mathsf{refl\text{-}Tm} && : \{\Gamma : \mathsf{Con}\}(A : \mathsf{Ty}\ \Gamma) \\
& && \to \mathsf{Tm}\ (\mathsf{rpl\text{-}T}\ \{\Delta = \mathsf{x{:}*}\}\ A\ (\mathsf{var\ v0 =h\ var\ v0})) \\
&\mathsf{refl\text{-}Tm}\ A && = \mathsf{rpl\text{-}tm}\ A\ \mathsf{refl*}
\end{aligned}
$$

Symmetry (inverse) It is defined similarly. Note that the intricate names of contexts, as in Ty x:*,y:*,α:x=y indicate their definitions which have been hidden. For instance we are assuming the definition: x:*,y:*,α:x=y = ε ,* , * , (var (vS v0) =h var v0)

```
sym*-Ty : Ty x:*,y:*,α:x=y
sym*-Ty = vY =h vX

sym*-Tm : Tm {x:*,y:*,α:x=y} sym*-Ty
sym*-Tm = Coh-Contr (ext c* v0)

sym-Tm : ∀ {Γ}(A : Ty Γ) → Tm (rpl-T A sym*-Ty)
sym-Tm A = rpl-tm A sym*-Tm
```

Trasitivity (composition) Note that each of these cells is defined by a different choice of the contractible context Δ.

```
trans*-Ty : Ty x:*,y:*,α:x=y,z:*,β:y=z
trans*-Ty = (vX +tm _ +tm _) =h vZ

trans*-Tm : Tm trans*-Ty
trans*-Tm = Coh-Contr (ext (ext c* v0) (vS v0))

trans-Tm : ∀ {Γ}(A : Ty Γ) → Tm (rpl-T A trans*-Ty)
trans-Tm A = rpl-tm A trans*-Tm
```

For each of reflexivity, symmetry and transitivity we can construct appropriate coherence 2-cells witnessing the groupoid axioms. The base case for variable contexts is proved simply using contractibility. We use substitution to define the application of the three basic terms we have defined above.

```
Tm-right-identity* : Tm {x:*,y:*,α:x=y}
   (trans*-Tm [ IdCm , vY , reflY ]tm =h vα)
Tm-right-identity* = Coh-Contr (ext c* v0)

Tm-left-identity* : Tm {x:*,y:*,α:x=y}
   (trans*-Tm [ ((IdCm ⊚ pr1 ⊚ pr1) , vX) ,
      reflX , vY , vα ]tm =h vα)
Tm-left-identity* = Coh-Contr (ext c* v0)

Tm-right-inverse* : Tm {x:*,y:*,α:x=y}
   (trans*-Tm [ (IdCm , vX) , sym*-Tm ]tm =h reflX)
Tm-right-inverse* = Coh-Contr (ext c* v0)

Tm-left-inverse* : Tm {x:*,y:*,α:x=y}
   (trans*-Tm [ ((• , vY) , vX , sym*-Tm , vY) , vα ]tm =h reflY)
Tm-left-inverse* = Coh-Contr (ext c* v0)

Tm-G-assoc* : Tm Ty-G-assoc*
Tm-G-assoc* = Coh-Contr (ext (ext (ext c* v0) (vS v0)) (vS v0))
```

Their general versions are defined using replacement. For instance, for associativity, we define:

```
Tm-G-assoc     : {Γ : Con}(A : Ty Γ)
               → Tm (rpl-T A Ty-G-assoc*)
Tm-G-assoc A   = rpl-tm A Tm-G-assoc*
```

3.3 Higher Structure

In this section we propose how also higher groupoid structure can be introduced in the syntactical framework. We use the more abstract language of category theory to communicate the gist of the construction leaving the tedious formalisation for future work. To this end note that contexts and context morphisms form a category up to definitional quality. Because equality of contexts is decidable we may assume UIP on context morphisms and we are therefore working in a honest 1-category where equality of arrows is definitional equality of context morphisms. This category will be denoted Con.

3.3.1 *Identities*

For each type of level $n \in \mathbf{N}$, we have defined in Section 3.1.2 a context called *span* which has $2n+1$ variables which except for the top level, n, there are two variables on each level whose type is the equality type of the two variables on the level below, except for the bottom-level variables which are of type $*$. We call denote a span of any type of level n, S_n. Note that all such spans are isomorphic.

In each case we call the last variable the *peak*. Note that each S_n is contractible because it is a suspension of a contractible context. We call the proof of contractibility of S_n is-contr S_n.

In each S_n define the type σ_n as $x_{2n-2} =_{\mathsf{h}} x_{2n-1}$. It is the type of the top variable. We are going to show that the following

$$S_0 \underset{t_0}{\overset{s_0}{\leftleftarrows}} \!\xrightarrow{i_0}\! S_1 \underset{t_1}{\overset{s_1}{\leftleftarrows}} \!\xrightarrow{i_1}\! S_2 \cdots S_n \underset{t_n}{\overset{s_n}{\leftleftarrows}} \!\xrightarrow{i_n}\! S_{n+1} \cdots$$

is a *reflexive globular object* in Con. I.e. we define morphisms s_n, t_n, i_n between spans that it satisfy the following usual *globular identities*:

$$\begin{aligned} s_n t_{n+1} &= s_n s_{n+1} \\ t_n t_{n+1} &= t_n s_{n+1} \end{aligned} \tag{1}$$

together with

$$s_n i_n \;=\; \mathsf{id} \;=\; t_n i_n \tag{2}$$

To this end, for each n, define context morphisms $s_n, t_n : S_{n+1} \Rightarrow S_n$ by the substitutions

$$\begin{aligned} s_n \;=\; & \quad x_k \mapsfrom x_k & k < 2n \\ & x_{2n} \mapsfrom x_{2n} \\ t_n \;=\; & \quad x_k \mapsfrom x_k & k < 2n \\ & x_{2n} \mapsfrom 2n+1 \end{aligned}$$

In words, s_n forgets the last two variables of S_{n+1} and t_n forgets the peak and its domain. It's easy to check that s and t indeed satisfy (1).

In order to define $i_n : S_n \Rightarrow S_{n+1}$, we must consider stalks (see Section 3.1), which are contexts, hereby denoted $\overline{S_n}$, which are like S_n without the last variable, together with types $\overline{\sigma}_n$ which are like σ_n but considered in the smaller context.

For each n there is a context morphism $\overline{i_n} : S_n \Rightarrow \overline{S}_{n+1}$ defined by

$$\begin{aligned} \overline{i_n} \;=\; & \quad x_k \mapsfrom x_k & k \leq 2n \\ & x_{2n+1} \mapsfrom x_{2n} \end{aligned}$$

The substitution of $\overline{\sigma}_{n+1}$ along $\overline{i_n}$, $\overline{i_n}[\overline{\sigma}_{n+1}]_{\mathsf{T}}$, is equal to $x_{2n} =_{\mathsf{h}} x_{2n}$ in S_n. So in order to extend $\overline{i_n}$ to $i_n : S_n \Rightarrow S_{n+1}$ we must define a term in $\overline{i_n}[\overline{\sigma}_{n+1}]_{\mathsf{T}}$. We can readily do that by coh:

$$i_n \;=\; \overline{i_n}\,,\, \mathsf{coh}\ (\mathsf{IdCm}\, \mathsf{S_n})\ (x_{2n} =_{\mathsf{h}} x_{2n})\ (\mathsf{is\text{-}contr}\ S_n)$$

It's easy to check that i_n satisfies (2).

For each n consider the chain

$$i_0^n \quad \equiv \quad S_0 \xrightarrow{i_0} S_1 \xrightarrow{i_1} S_2 \longrightarrow \cdots \xrightarrow{i_{n-1}} S_n$$

The substitution $\sigma_n[i_0]_{\mathsf{T}} \cdots [i_{n-2}]_{\mathsf{T}}[i_{n-1}]_{\mathsf{T}} \equiv \sigma_n[i_0^n]_{\mathsf{T}}$ is a type, λ_n, in S_0. We call λ_n the *n-iterated loop type* on x_0. The term $S_0 \vdash (\mathsf{var}\ x_{2n})[i_0^n]_{\mathsf{tm}} : \lambda_n$ is the iterated identity term on x_0.

3.3.2 Composition

For $m > n$ write s_n^m, t_n^m for $m-n$ iteration of s and t, respectively. Explicitly:

$$\begin{aligned} s_n^m &= s_{m-1} \cdots s_n &&: S_m \Rightarrow S_n \\ t_n^m &= t_{m-1} \cdots t_n &&: S_m \Rightarrow S_n \end{aligned}$$

For each $n \in \mathbb{N}$ define context Y_n by the pullback in:

$$\begin{tikzcd} S_0 & S_n \arrow[l, "t_0^n"'] \\ S_n \arrow[u, "s_0^n"] & Y_n \arrow[l, "r_n"] \arrow[u, "l_n"'] \arrow[ul, phantom, "\ulcorner", very near start] \end{tikzcd}$$

By definition of pullbacks, Y_n looks like a pair of spans S_n together with the proviso that the variable 1 of one is always equal to variable 0 of the other. I.e. Y_n has the shape of two spans pasted target-to-source at level 0. It is easy to check that this is indeed a pullback.

By the globular identities the two outer squares in the diagram below commute and by the universal property of the pullback imply a pair of mediating arrows as indicated.

$$\begin{tikzcd} S_0 & S_n \arrow[l, "t_0^n"'] & S_{n+1} \arrow[l, "s_n"'] \\ S_n \arrow[u, "s_0^n"] & Y_n \arrow[l, "r_n"] \arrow[u, "l_n"'] \arrow[ul, phantom, "\ulcorner", very near start] & \\ S_{n+1} \arrow[u, "s_n"] & & Y_{n+1} \arrow[ll, "r_{n+1}"] \arrow[uu, "l_{n+1}"'] \arrow[ul, dashed, "\langle s,s \rangle_n"'] \arrow[ul, phantom, "\ulcorner", very near start] \end{tikzcd}$$

Similarly we obtain an arrow $Rr_n : Y_{n+1} \longrightarrow Y_n$. The morphisms l_n and r_n provide projections onto the left and right span of Y_n respectively. The mediating arrows $\langle s,s \rangle_n$ and $\langle t,t \rangle_n$ provide projections out of Y_{n+1} onto the join of the sources and targets of the left and right parts respectively.

In order to define composition we define for each n a third morphism $c_n : Y_n \Rightarrow S_n$ with the property that both the s-squares and t-squares below commute.

$$\begin{tikzcd} S_n & S_{n+1} \arrow[l, shift right, "s_n"'] \arrow[l, shift left, "t_n"] \\ Y_n \arrow[u, "c_n"] & Y_{n+1} \arrow[l, shift right, "\langle s,s \rangle_n"'] \arrow[l, shift left, "\langle t,t \rangle_n"] \arrow[u, "c_{n+1}"'] \end{tikzcd} \tag{3}$$

The commutativity of (3) expresses the fact that the source of a composition is a composition of sources and the target of a composition is a composition of target.

It follows from all of this that for a context Γ and a pair of morphisms $a, b : \Gamma \Rightarrow S_n$, there is a context morphism $c\langle a, b \rangle : \Gamma \Rightarrow S_n$ from $s_0^n a$ to $t_0^n b$ which is the composition of a and b.

4. SEMANTICS

4.1 Globular Types

To interpret the syntax, we need globular types [1] . Globular types are defined coinductively as follows.

```
record Glob : Set₁ where
  constructor _||_
  field
    |_|   : Set
    hom   : |_| → |_| → ∞ Glob
```

If all the object types are indeed sets, i.e. uniqueness of identity types holds, we call this a globular set.

As an example, we could construct the identity globular type called Idw.

```
Idω      : (A : Set) → Glob
Idω A    = A || (λ a b → ♯ Idω (a ≡ b))
```

Note that this is usually not a globular set.

Given a globular set G, we can interpret the syntactic objects.

The record definition also require some semantic weakening and semantic substitution laws. The semantic weakening rules tell us how to deal with the weakening inside interpretation.

```
record Semantic (G : Glob) : Set₁ where
  field
    ⟦_⟧C   : Con → Set
    ⟦_⟧T   : ∀{Γ} → Ty Γ → ⟦ Γ ⟧C → Glob
    ⟦_⟧tm  : ∀{Γ A} → Tm A → (γ : ⟦ Γ ⟧C) → | ⟦ A ⟧T γ |
    ⟦_⟧cm  : ∀{Γ Δ} → Γ ⇒ Δ → ⟦ Γ ⟧C → ⟦ Δ ⟧C
    π      : ∀{Γ A} → Var A → (γ : ⟦ Γ ⟧C) → | ⟦ A ⟧T γ |
```

π provides the projection of the semantic variable out of a semantic context.

Following are the computation laws for the interpretations of contexts and types.

```
⟦_⟧C-β1   : ⟦ ε ⟧C ≡ ⊤
⟦_⟧C-β2   : ∀ {Γ A} → ⟦ Γ , A ⟧C ≡
  Σ ⟦ Γ ⟧C (λ γ   → | ⟦ A ⟧T γ |)

⟦_⟧T-β1   : ∀{Γ}{γ : ⟦ Γ ⟧C} → ⟦ * ⟧T γ ≡ G
⟦_⟧T-β2   : ∀{Γ A u v}{γ : ⟦ Γ ⟧C}
            → ⟦ u =h v ⟧T γ ≡
              ♭ (hom (⟦ A ⟧T γ) (⟦ u ⟧tm γ) (⟦ v ⟧tm γ))
```

The semantic substitution properties are essential,

[1] Even though we use the Agda |Set|, this isn't necessarily a set in the sense of Homotopy Type Theory.

$$\mathsf{semSb\text{-}T} \;:\; \forall\, \{\Gamma\ \Delta\}(A : \mathsf{Ty}\ \Delta)(\delta : \Gamma \Rightarrow \Delta)(\gamma : [\![\ \Gamma\ ]\!]C) \\ \to [\![\ A\ [\ \delta\ ]\mathsf{T}\ ]\!]T\ \gamma \equiv [\![\ A\ ]\!]T\ ([\![\ \delta\ ]\!]cm\ \gamma)$$

$$\mathsf{semSb\text{-}tm} : \forall\{\Gamma\ \Delta\}\{A : \mathsf{Ty}\ \Delta\}(a : \mathsf{Tm}\ A)(\delta : \Gamma \Rightarrow \Delta) \\ (\gamma : [\![\ \Gamma\ ]\!]C) \\ \to \mathsf{subst}\ |\_|\ (semSb\text{-}T\ A\ \delta\ \gamma)\ ([\![\ a\ [\ \delta\ ]\mathsf{tm}\ ]\!]tm\ \gamma) \\ \equiv [\![\ a\ ]\!]tm\ ([\![\ \delta\ ]\!]cm\ \gamma)$$

$$\mathsf{semSb\text{-}cm} : \forall\, \{\Gamma\ \Delta\ \Theta\}(\gamma : [\![\ \Gamma\ ]\!]C)(\delta : \Gamma \Rightarrow \Delta)(\vartheta : \Delta \Rightarrow \Theta) \\ \to [\![\ \vartheta \circledcirc \delta\ ]\!]cm\ \gamma \equiv [\![\ \vartheta\ ]\!]cm\ ([\![\ \delta\ ]\!]cm\ \gamma)$$

Since the computation laws for the interpretations of terms and context morphisms are well typed up to these properties.

$$[\![\_]\!]\mathsf{tm}\text{-}\beta 1 \;:\; \forall\{\Gamma\ A\}\{x : \mathsf{Var}\ A\}\{\gamma : [\![\ \Gamma\ ]\!]C\} \\ \to [\![\ \mathsf{var}\ x\ ]\!]tm\ \gamma \equiv \pi\ x\ \gamma$$

$$[\![\_]\!]\mathsf{cm}\text{-}\beta 1 \;:\; \forall\{\Gamma\}\{\gamma : [\![\ \Gamma\ ]\!]C\} \\ \to [\![\ \bullet\ ]\!]cm\ \gamma \equiv \mathsf{coerce}\ [\![\_]\!]C\text{-}\beta 1\ \mathsf{tt}$$

$$[\![\_]\!]\mathsf{cm}\text{-}\beta 2 \;:\; \forall\{\Gamma\ \Delta\}\{A : \mathsf{Ty}\ \Delta\}\{\delta : \Gamma \Rightarrow \Delta\}\{\gamma : [\![\ \Gamma\ ]\!]C\} \\ \{a : \mathsf{Tm}\ (A\ [\ \delta\ ]\mathsf{T})\} \to [\![\ \delta\ ,\ a\ ]\!]cm\ \gamma \\ \equiv \mathsf{coerce}\ [\![\_]\!]C\text{-}\beta 2\ (([\![\ \delta\ ]\!]cm\ \gamma)\ , \\ \mathsf{subst}\ |\_|\ (semSb\text{-}T\ A\ \delta\ \gamma)\ ([\![\ a\ ]\!]tm\ \gamma))$$

The semantic weakening properties should actually be deriavable since weakening is equivalent to projection substitution.

$$\mathsf{semWk\text{-}T} \;:\; \forall\, \{\Gamma\ A\ B\}(\gamma : [\![\ \Gamma\ ]\!]C)(v : |\ [\![\ B\ ]\!]T\ \gamma\ |) \\ \to [\![\ A +\mathsf{T}\ B\ ]\!]T\ (\mathsf{coerce}\ [\![\_]\!]C\text{-}\beta 2\ (\gamma\ ,\ v)) \equiv \\ [\![\ A\ ]\!]T\ \gamma$$

$$\mathsf{semWk\text{-}cm} \;:\; \forall\, \{\Gamma\ \Delta\ B\}\{\gamma : [\![\ \Gamma\ ]\!]C\}\{v : |\ [\![\ B\ ]\!]T\ \gamma\ |\} \\ \to (\delta : \Gamma \Rightarrow \Delta) \to [\![\ \delta +\mathsf{S}\ B\ ]\!]cm \\ (\mathsf{coerce}\ [\![\_]\!]C\text{-}\beta 2\ (\gamma\ ,\ v)) \equiv [\![\ \delta\ ]\!]cm\ \gamma$$

$$\mathsf{semWk\text{-}tm} : \forall\, \{\Gamma\ A\ B\}(\gamma : [\![\ \Gamma\ ]\!]C)(v : |\ [\![\ B\ ]\!]T\ \gamma\ |) \\ \to (a : \mathsf{Tm}\ A) \to \mathsf{subst}\ |\_|\ (semWk\text{-}T\ \gamma\ v) \\ ([\![\ a +\mathsf{tm}\ B\ ]\!]tm\ (\mathsf{coerce}\ [\![\_]\!]C\text{-}\beta 2\ (\gamma\ ,\ v))) \\ \equiv ([\![\ a\ ]\!]tm\ \gamma)$$

Here we declare them as properties because they are essential for the computation laws of function π.

$$\pi\text{-}\beta 1 \;:\; \forall\{\Gamma\ A\}(\gamma : [\![\ \Gamma\ ]\!]C)(v : |\ [\![\ A\ ]\!]T\ \gamma\ |) \\ \to \mathsf{subst}\ |\_|\ (semWk\text{-}T\ \gamma\ v) \\ (\pi\ \mathsf{v0}\ (\mathsf{coerce}\ [\![\_]\!]C\text{-}\beta 2\ (\gamma\ ,\ v))) \equiv v$$

$$\pi\text{-}\beta 2 \;:\; \forall\{\Gamma\ A\ B\}(x : \mathsf{Var}\ A)(\gamma : [\![\ \Gamma\ ]\!]C)(v : |\ [\![\ B\ ]\!]T\ \gamma\ |) \\ \to \mathsf{subst}\ |\_|\ (semWk\text{-}T\ \gamma\ v)\ (\pi\ (\mathsf{vS}\ \{\Gamma\}\ \{A\}\ \{B\}\ x) \\ (\mathsf{coerce}\ [\![\_]\!]C\text{-}\beta 2\ (\gamma\ ,\ v))) \equiv \pi\ x\ \gamma$$

The only part of the semantics where we have any freedom is the interpretation of the coherence constants:

$$[\![\mathsf{coh}]\!] : \forall\{\Theta\} \to \mathsf{isContr}\ \Theta \to (A : \mathsf{Ty}\ \Theta) \\ \to (\vartheta : [\![\ \Theta\ ]\!]C) \to |\ [\![\ A\ ]\!]T\ \vartheta\ |$$

However, we also need to require that the coherence constants are well behaved wrt to substitution which in turn relies on the interpretation of all terms. To address this we state the required properties in a redundant form because the correctness for any other part of the syntax follows from the defining equations we have already stated.There seems to be no way to avoid this.

If the underlying globular type is not a globular set we need to add coherence laws, which is not very well understood. On the other hand, restricting ourselves to globular sets means that our prime examle $\mathsf{Id}\omega$ is not an instance anymore. We should still be able to construct non-trivial globular sets, e.g. by encoding basic topological notions and defining higher homotopies as in a classical framework. However, we don't currently know a simple definition of a globular set which is a weak ω-groupoid. One possibility would be to use the syntax of type theory with equality types. Indeed, we believe that this would be an alternative way to formalize weak ω-groupoids.

5. CONCLUSION

In this paper, we present an implementation of weak ω-groupoids following Brunerie's work. Briefly speaking, we define the syntax of the type theory $\mathcal{T}_{\infty-groupoid}$, then a weak ω-groupoid is a globular set with the interpretation of the syntax. To overcome some technical problems, we use heterogeneous equality for terms, some auxiliary functions and loop context in all implementation. We construct the identity morphisms and verify some groupoid laws in the syntactic framework. The suspensions for all sorts of objects are also defined for other later constructions.

There is still a lot of work to do within the syntactic framework. For instance, we would like to investigate the relation between the $\mathcal{T}_{\infty-groupoid}$ and a Type Theory with equality types and J eliminator which is called $\mathcal{T}_{eq}$. One direction is to simulate the J eliminator syntactically in $\mathcal{T}_{\infty-groupoid}$ as we mentioned before, the other direction is to derive J using *coh* if we can prove that the $\mathcal{T}_{eq}$ is a weak ω-groupoid. The syntax could be simplified by adopting categories with families. An alternative may be to use higher inductive types directly to formalize the syntax of type theory.

We would like to formalise a proof of that $\mathsf{Id}\omega$ is a weak ω-groupoid, but the base set in a globular set is an h-set which is incompatible with $\mathsf{Id}\omega$. Perhaps we could solve the problem by instead proving a syntactic result, namely that the theory we have presented here and the theory of equality types with J are equivalence. Finally, to model the Type Theory with weak ω-groupoids and to eliminate the univalence axiom would be the most challenging task in the future.

6. REFERENCES

[1] Thorsten Altenkirch. Extensional Equality in Intensional Type Theory. In *14th Symposium on Logic in Computer Science*, pages 412 – 420, 1999.

[2] Thorsten Altenkirch, Conor McBride, and Wouter Swierstra. Observational equality, now! In *PLPV '07: Proceedings of the 2007 workshop on Programming languages meets program verification*, pages 57–68, New York, NY, USA, 2007. ACM.

[3] Thorsten Altenkirch and Ondrej Rypacek. A syntactical Approach to Weak ω-groupoids. In *Computer Science Logic (CSL'12) - 26th International Workshop/21st Annual Conference of the EACSL, CSL 2012*, 2012.

[4] D. Ara. On the homotopy theory of Grothendieck ∞-groupoids. *ArXiv e-prints*, June 2012.

[5] Guillaume Brunerie. Syntactic Grothendieck weak

∞-groupoids, 2013.

[6] Alexander Grothendieck. Pursuing Stacks. 1983. Manuscript.

[7] Michael Hedberg. A coherence theorem for Martin-Löf's type theory. 1998.

[8] G. Maltsiniotis. Grothendieck ∞-groupoids, and still another definition of ∞-categories. *ArXiv e-prints*, September 2010.

[9] The Univalent Foundations Program. *Homotopy type theory: Univalent foundations of mathematics*. 2013.